

## Chapitre 8 – Solutions des problèmes

### 1. Le congrès de l'Association internationale de marketing.

(a) La figure ci-dessous indique comment nous avons procédé. Tout d'abord, la formule =ALEA() de la colonne B commande à Excel de générer 1000 nombres pseudo-aléatoires entre 0 et 1. Ces derniers sont ensuite convertis en valeurs pseudo-aléatoires de la variable  $X$ ; les paramètres  $N Moy$  et  $N Écart T$  de la formule en C5 représentent la valeur moyenne 4500 et l'écart type 250 de la variable normale  $X$ . Enfin, les colonnes D à M donnent les coûts qu'aurait encourus l'Association si le nombre de réservations avait été fixé à 4350, 4370, 4390, etc. La formule de la plage D5:M1004 couvre deux cas d'espèce.

- Considérons d'abord la cellule D5. Le nombre  $x = 4517$  de congressistes (voir cellule C5) est supérieur au nombre  $r = 4350$  de réservations (D4). L'Association doit payer 3 nuitées pour chacune des chambres supplémentaires requises par les  $x - r = 167$  congressistes en excédent. Le coût à reporter en D5 s'élève donc à 15 030 dollars :

$$3 \times (80 - 50) \times (x - r) = 3 \times 30 \times 167 = 15\,030.$$

De même, on retrouve en E5 un coût de 13 230 dollars :

$$3 \times (80 - 50) \times (x - r) = 3 \times 30 \times (4517 - 4370) = 3 \times 30 \times 147 = 13\,230.$$

- Par contre, dans le cas traité en M5, le nombre  $x = 4517$  de congressistes est inférieur au nombre  $r = 4530$  de réservations (M4). L'Association doit payer les  $r - x = 13$  chambres en trop, soit 1 950 dollars au total :

$$3 \times 50 \times (r - x) = 3 \times 50 \times (4530 - 4517) = 3 \times 50 \times 13 = 1\,950.$$

Fichier : MOG8-01.xlsx										Feuilles : SimulA et Fig1			
	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Coût (en \$) encourus par l'Association pour les chambres en moins ou en trop, selon le nombre de chambres retenues												
2													
3	Nombre de chambres retenues												
4	N°	NPA	X	4 350	4 370	4 390	4 410	4 430	4 450	4 470	4 490	4 510	4 530
5	1	0,527	4517	15 030	13 230	11 430	9 630	7 830	6 030	4 230	2 430	630	1 950
6	2	0,616	4574	20 160	18 360	16 560	14 760	12 960	11 160	9 360	7 560	5 760	3 960
7	3	0,719	4645	26 550	24 750	22 950	21 150	19 350	17 550	15 750	13 950	12 150	10 350
8	4	0,056	4102	37 200	40 200	43 200	46 200	49 200	52 200	55 200	58 200	61 200	64 200
9	5	0,485	4490	12 600	10 800	9 000	7 200	5 400	3 600	1 800	0	3 000	6 000
1002	998	0,558	4537	16 830	15 030	13 230	11 430	9 630	7 830	6 030	4 230	2 430	630
1003	999	0,000	3580	115 500	118 500	121 500	124 500	127 500	130 500	133 500	136 500	139 500	142 500
1004	1000	0,537	4523	15 570	13 770	11 970	10 170	8 370	6 570	4 770	2 970	1 170	1 050

Cellule	Formule	Copiée dans :
B5	=ALEA()	B6:B1004
C5	=ARRONDI(LOI.NORMALE.INVERSE.N(B4;NMoy;NÉcartT);0)	C6:C1004
D5	=3*SI(D\$4>=\$C5;50*(D\$4-\$C5);30*(C5-D\$4))	D5:M1004

La figure suivante résume les résultats obtenus. On constate (voir ligne 8) que la meilleure stratégie semble de réserver environ 4410 chambres.

Fichier : MOG8-01.xlsx		Feuilles : RésultatsA et Fig2									
	A	B	C	D	E	F	G	H	I	J	K
1	Coût (en \$) encourus par l'Association pour les chambres en moins ou en trop, selon le nombre de chambres retenues										
2											
3		Nombre de chambres retenues									
4		4 350	4 370	4 390	4 410	4 430	4 450	4 470	4 490	4 510	4 530
5	Minimum	90	0	150	0	0	0	0	0	0	0
6	Médiane	19 800	19 170	18 630	18 735	18 270	18 450	18 450	19 125	19 650	20 175
7	Maximum	124 200	127 200	130 200	133 200	136 200	139 200	142 200	145 200	148 200	151 200
8	Moyenne $\mu$	23 384	23 028	22 835	22 793	22 884	23 121	23 501	24 050	24 735	25 604
9	Intervalle à 90% pour $\mu$ : B.Inf	22 456	22 092	21 888	21 831	21 900	22 109	22 456	22 969	23 612	24 439
10	Intervalle à 90% pour $\mu$ : B.Sup	24 313	23 964	23 781	23 755	23 869	24 133	24 546	25 132	25 858	26 769
Cellule	Formule										Copiée dans :
B5	=MIN(SimulA!D\$5:D\$10044)										C5:K5
B6	=MEDIANE(SimulA!D\$5:D\$1004)										C6:K6
B7	=MAX(SimulA!D\$5:D\$1004)										C7:K7
B8	=MOYENNE(SimulA!D\$5:D\$1004)										C8:K8
B9	=B8-1,645*ECARTYPE(SimulA!D\$5:D\$1004)/1000^0,5										C9:K9
B10	=B8+1,645*ECARTYPE(SimulA!D\$5:D\$1004)/1000^0,5										C10:K10

(b) On reprend les calculs de la feuille SimulA, mais en reportant en C5 la formule suivante :

$$=ARRONDI(SI(B5<=Tk;TMin+(TMax-TMin)*RACINE(Tk*B5);TMax-(TMax-TMin)*RACINE((1-Tk)*(1-B5)));0)$$

Les paramètres TMin et TMax représentent les valeurs minimale et maximale de la variable triangulaire X. Enfin, Tk correspond à la proportion de l'intervalle [TMin; TMax] située à la gauche de la valeur modale TMod et se calcule selon la formule donnée à la page 441 :

$$Tk = \frac{TMod - TMin}{TMax - TMin} = \frac{4000 - 2300}{7000 - 2300} = 0,362.$$

La formule ci-dessus s'obtient de la formule correspondante de la question (a) en remplaçant la fonction LOI.NORMALE.INVERSE.N d'Excel par la fonction inverse d'une loi triangulaire (voir la formule de la cellule G6 de la figure 8.19).

La figure suivante résume les résultats obtenus. On constate (voir ligne 8) que la meilleure stratégie semble de réserver entre 4010 et 4030 chambres.

Fichier : MOG8-01.xlsx		Feuilles : RésultatsB et Fig3									
	A	B	C	D	E	F	G	H	I	J	K
1	Coût (en \$) encourus par l'Association pour les chambres en moins ou en trop, selon le nombre de chambres retenues										
2											
3		Nombre de chambres retenues									
4		3 950	3 970	3 990	4 010	4 030	4 050	4 070	4 090	4 110	4 130
5	Minimum	0	0	90	150	0	0	0	90	150	270
6	Médiane	77 925	78 555	78 570	78 465	78 345	77 580	77 490	76 950	76 365	77 250
7	Maximum	266 580	264 780	262 980	261 180	259 380	261 750	264 750	267 750	270 750	273 750
8	Moyenne $\mu$	88 633	88 505	88 408	88 342	88 347	88 416	88 528	88 665	88 841	89 056
9	Intervalle à 90% pour $\mu$ : B.Inf	85 314	85 188	85 090	85 022	85 025	85 092	85 201	85 332	85 501	85 707
10	Intervalle à 90% pour $\mu$ : B.Sup	91 952	91 822	91 725	91 663	91 669	91 739	91 855	91 997	92 181	92 405

**Note 1.** Dans la figure 1 ci-dessus, nous avons utilisé les mêmes nombres pseudo-aléatoires pour comparer les différentes stratégies qui s'offrent à l'Association : en effet, les coûts calculés dans les colonnes D à M sont tous basés sur les valeurs de la variable  $X$  qui se trouvent dans la colonne C. Nous avons procédé de la même façon dans la question (b). Cette approche découle de la règle, énoncée à la page 419, qui recommande d'apparier, autant que possible, les données employées pour comparer des groupes. Cependant, les analyses effectuées en (a) et en (b) sont indépendantes, et il ne nous a pas semblé nécessaire de reprendre en (b) les mêmes nombres pseudo-aléatoires qu'en (a).

**Note 2.** Il était difficile *a priori* de deviner la valeur approximative de nombre optimal de réservations. Dans la question (a), nous avons effectué un premier essai en comparant les coûts lorsque le nombre  $r$  de réservations variait de 4000 à 4900 par pas de 100. Nous avons alors constaté que le coût minimal semblait être obtenu quand  $r$  prenait une valeur près de 4400. Nous avons donc raffiné notre analyse en procédant à un deuxième essai dans lequel  $r$  variait de 4350 à 4450 par pas de 10. De même, en (b), nous avons d'abord fait varier  $r$  de 3000 à 4800 par pas de 200.

## 2. Une chaîne de montage en nanotechnologie.

La figure ci-dessous indique comment nous avons procédé. Nous commenterons d'abord les formules des colonnes B à E qui servent à répondre à la question (a). Nous expliquerons celle de la dernière colonne dans la réponse à la question (b).

Fichier : MOG8-02.xlsx				Feuilles : Simul et Fig1		
	A	B	C	D	E	F
1	N°	$X_A$	$X_B$	$X_C$	Y	Rev. net
2	1	62	62	52	52	172 850 \$
3	2	51	64	55	51	171 520 \$
4	3	46	58	54	46	149 280 \$
5	4	51	52	63	51	168 640 \$
6	5	56	54	62	54	183 050 \$
1000	999	47	66	51	47	153 760 \$
1001	1000	47	53	49	47	163 440 \$
Cellule	Formule					Copiée dans :
B2	=AMin+ARRONDI(ALEA()*(AMax-AMin+1)-0,5;0)					B3:B1001
C2	=BMin+ARRONDI(ALEA()*(BMax-BMin+1)-0,5;0)					C3:C1001
D2	=CMin+ARRONDI(ALEA()*(CMax-CMin+1)-0,5;0)					D3:D1001
E2	=MIN(B2;C2;D2)					E3:E1001
F2	=RevU*E2-CtA*MAX(B2-E2;0)-CtB*MAX(C2-E2;0)-CtC*MAX(D2-E2;0)					F3:F1001

(a) Les formules des colonnes B à D génèrent des nombres pseudo-aléatoires obéissant à des lois uniformes discrètes. Analysons plus en détail celle de B. Les paramètres  $AMin = 45$  et  $AMax = 65$  représentent les valeurs extrêmes de la variable discrète  $X_A$ , le nombre de composants produites une journée donnée par le laboratoire A.

Observons d'abord que la variable discrète  $X_A$  peut prendre 21 valeurs, soit 45, 46, 47, ..., 65. Nous nous fixerons comme premier objectif de générer de façon équiprobable des nombres entiers pseudo-aléatoires entre 0 et 20. Le terme  $ALEA()$  de la formule commande de générer des nombres réels entre 0 et 1. En multipliant par  $(AMax - AMin + 1) = 21$ , puis en retranchant 0,5, on obtient un nombre réel entre  $-0,5$  et  $20,5$ , que l'on transforme ensuite en nombre entier à l'aide de la fonction  $ARRONDI$  d'Excel. Noter que le terme  $-0,5$  est nécessaire, sinon 22 valeurs seraient possibles, de 0 à 21 inclusivement (cette dernière valeur surviendrait lorsque la fonction  $ALEA()$  renvoie un nombre supérieur à  $1 - 1 / (2 \times 21) = 0,9762$ ).

Il suffit ensuite d'additionner la constante  $AMin$  pour obtenir le nombre de composantes produites pendant la journée par le laboratoire A, qui sera un nombre entier entre  $AMin = 45$  et  $AMax = 65$ .

Enfin, la formule de la colonne E est la définition de la variable  $Y$ .

Le tableau ci-dessous résume les résultats obtenus. Les formules utilisées sont semblables à celles de la figure 2 du problème 1 et ne sont pas reproduites ici. On constate que  $Y$  varie de 40 à 65, avec la valeur 47 comme médiane. De plus, le nombre moyen de nano-objets assemblés en une journée s'élève à 48,32 environ.

Fichier : MOG8-02.xlsx		Feuilles : Résultats et Fig2				
	A	B	C	D	E	F
1		$X_A$	$X_B$	$X_C$	$Y$	Rev. net
2	Minimum	45	42	40	40	105 955 \$
3	5 <sup>e</sup> centile	45	43	41	41	121 149 \$
4	Médiane	55	58	54	47	154 220 \$
5	95 <sup>e</sup> centile	64	71	67	59	203 231 \$
6	Maximum	65	72	68	65	230 000 \$
7	Moyenne $\mu$	54,7	57,4	53,9	48,32	156 603 \$
8	Écart type	6,0	8,8	8,4	5,38	24 805 \$
9	Intervalle à 90% pour $\mu$ : B.Inf	54,4	56,9	53,4	48,04	155 312 \$
10	Intervalle à 90% pour $\mu$ : B.Sup	55,1	57,8	54,3	48,60	157 893 \$

(b) Le revenu quotidien net de la chaîne de montage est calculé dans la colonne F de la figure 1. Les noms  $RevU$ ,  $CtA$ ,  $CtB$  et  $CtC$  font référence à des cellules de la feuille Données et représentent les paramètres de notre modèle :


$$RevU = 3600 \text{ et } CtA = 875 \text{ et } CtB = 560 \text{ et } CtC = 1200.$$

Le premier terme de la formule en F2 donne le profit brut de la journée : il s'agit tout simplement de multiplier le profit unitaire par le nombre de nano-objets assemblés. De ce montant, il faut retrancher les coûts de mise au rebut. Par exemple, pour les deux premières journées considérées à la figure 2, le revenu net s'élève à 172 850 \$ et à 171 520 \$ respectivement :

$$(3600 \times 52) - (875 \times 10) - (560 \times 10) - (1200 \times 0) = 172\,850$$

$$(3600 \times 51) - (875 \times 0) - (560 \times 13) - (1200 \times 4) = 171\,520.$$

La colonne F de la figure 2 donne les principales caractéristiques de la distribution du revenu net pour les 1000 journées de l'échantillon. On constate que la chaîne procure un profit quotidien moyen de 156 603 \$, mais que le gain net varie considérablement d'une journée à l'autre, s'étalant d'un minimum de 105 955 \$ à un maximum de 203 000 \$. L'histogramme ci-dessous décrit la distribution du revenu net (en k\$) dans l'échantillon considéré aux figures 1 et 2.



**Note 1.** Nous indiquons brièvement comment nous avons obtenu cet histogramme.

- Étape préliminaire (non essentielle) : nous avons créé une feuille Hist-RevN et nous avons copié dans la plage A2:A1001 les revenus nets de la colonne F de la feuille Fig2; nous avons ensuite divisé ces nombres par 1000, afin d'alléger les étiquettes de l'axe horizontal.
- Construction proprement dite du graphique : cliquer sur le menu Données, puis sur l'option Utilitaire d'analyse et enfin sur Histogramme. Pour l'exemple traité ici, la boîte Histogramme a été complétée de la façon suivante :
  - Plage d'entrée : A2:A1001
  - Plage des classes : D2:D14
  - Plage de sortie : G1:L10
  - Cocher Représentation graphique.
- Mise en forme fine du graphique (non essentielle) : nous avons modifié le titre du graphique, ainsi que celui de l'axe vertical; nous avons supprimé le titre de l'axe horizontal et reporté «230» en G15; pour afficher les effectifs des classes, nous avons coché droit sur l'un des rectangles et avons cliqué sur Ajouter des étiquettes de données; de même, pour que les classes soient contiguës, nous avons coché droit sur l'un des rectangles et nous avons cliqué sur Mettre en forme une série de données, puis sur Options des séries et enfin, dans Largeur de l'intervalle, nous avons déplacé le curseur à Pas d'intervalle.

**Note 2.** Il faut ici utiliser des nombres pseudo-aléatoires différents dans les colonnes B, C et D de la figure 1, car les quantités fabriquées par les trois laboratoires sont des variables aléatoires indépendantes.

### 3. Les télescopes de Celistron.

(a) La figure ci-dessous indique comment simuler la vente de 1000 appareils. Tout d'abord, en colonne B, on génère 1000 nombres pseudo-aléatoires obéissant à une loi normale de moyenne 4 et d'écart type 1,5 (les noms MoyenneX et EcartTX de la formule en B3 renvoient à des cellules de la feuille Données contenant les valeurs 4 et 1,5).

La plage B3:B1002 donne les durées de vie des 1000 télescopes de notre échantillon. Considérons le 1000<sup>e</sup> et dernier cas de la figure (voir ligne 1002). Cet appareil fonctionnera correctement pendant seulement 1,38 année, ce qui est inférieur aux 2 ans de la garantie, et Celistron devra le remplacer. Le nouveau télescope aura lui aussi une durée de vie aléatoire obéissant à une loi normale  $N(4; 1,5^2)$ . Il se trouve (voir cellule C1002) qu'il tombera en panne après 1,92 année. Celistron fournira donc au client un 3<sup>e</sup> télescope qui durera 4,64 années, soit nettement plus que la garantie. En E1002, on obtiendra donc une réponse positive quand on testera si le contenu de D1002 est égal ou supérieur à 2; par conséquent, Excel va renvoyer une cellule vide en E1002.

La cellule C1003 compte combien de cellules ne sont pas vides dans la plage C3:C1002, c'est-à-dire combien, parmi les 1000 télescopes vendus, devront être remplacés une première fois par Celistron. De même, D1003 combien de télescopes de remplacement tomberont en panne prématurément et devront eux aussi être remplacés. Enfin, la cellule B1003 indique combien d'appareils au total Celistron remettra gratuitement aux clients :

$$69 + 10 + 3 + 2 + 0 = 84.$$

Fichier : MOG8-03.xlsx		Feuilles : S1000 et FigA						
	A	B	C	D	E	F	G	
1		Durée de vie du télescope de remplacement numéro						
2	N <sup>o</sup>	0	1	2	3	4	5	
3	1	4,63						
4	2	-0,20	1,54	1,77	0,92	5,40		
5	3	4,59						
997	995	3,38						
998	996	-1,08	6,45					
999	997	4,24						
1000	998	2,99						
1001	999	6,38						
1002	1000	1,38	1,92	4,64				
1003	RemplT	84	69	10	3	2	0	
Cellule		Formule					Copiée dans :	
B3	=LOI.NORMALE.INVERSE.N(ALEA());MoyenneX;EcartTX)					B4:B1002		
C3	=SI(B3>=Garantie;"";LOI.NORMALE.INVERSE.N(ALEA());MoyenneX;EcartTX))					C3:G1002		
C1003	=1000-NB.VIDE(C3:C1002)					D1003:G1003		
B1003	=SOMME(C1003:G1003)					-----		

Les feuilles RevN et Qa du fichier MOG8-03.xlsx analysent les conséquences monétaires des remplacements que Celistron devra effectuer. Dans le cas de l'échantillon de la figure ci-dessus, le revenu net par appareil vendu s'élève à 499,20 \$ :

$$\text{coût de la garantie} = 1678 \text{ \$/u} \times 84 \text{ u} = 140\,952 \text{ \$}$$

$$\text{coût de la garantie par vente} = 140\,952 \text{ \$} \div 1000 \text{ u} = 140,95 \text{ \$/u}$$

$$\text{revenu net par vente} = 2345 \text{ \$/u} - 1678 \text{ \$/u} - 140,95 \text{ \$/u} = 526,05 \text{ \$/u.}$$

(b) Dans un premier temps, on construit des feuilles S2000 et S5000 identiques à S1000, sauf que 2000 ou 5000 cas sont considérés au lieu de seulement 1000. Puis, on calcule le nombre observé de télescopes de remplacement par appareil vendu (voir les cellules C4, C5 et C6 de la figure qui suit). On répète l'opération 20 fois à l'aide de la commande Table de données... d'Excel. Les résultats obtenus sont reproduits dans la figure ci-dessous. On observe que les quotients  $\text{RemplT}/\text{Ventes}$  oscillent autour de 10%. De plus, la variabilité décroît lorsque le nombre  $N$  de ventes augmente : l'écart type des quotients (voir plage B26:D26) diminue de 103 à 79 et à 45 dix-millièmes lorsque  $N$  passe de 1000 à 2000, puis à 5000.

Fichier : MOG8-03.xlsx		Feuilles : Qb et FigB		
	A	B	C	D
3	No	1000	2000	5000
4		0,0900	0,0960	0,0990
5	1	0,1080	0,0975	0,0966
6	2	0,1090	0,0910	0,1050
7	3	0,0890	0,1080	0,0952
8	4	0,0930	0,0975	0,0982
9	5	0,1020	0,0835	0,1056
23	19	0,1140	0,1025	0,1034
24	20	0,0940	0,1030	0,0958
25	Moyenne	0,1004	0,0991	0,0996
26	Écart type	0,0103	0,0079	0,0045
Cellule	Formule	Copiée dans :		
B4	=S1000!B1003/1000	-----		
C4	=S2000!B2003/2000	-----		
D4	=S5000!B5003/5000	-----		
B25	=MOYENNE(B5:B24)	C25:D25		
B26	=ECARTYPE(B5:B24)	C26:D26		

**Note.** Nous nous intéressons ici à la valeur théorique du nombre attendu de télescopes de remplacement par appareil vendu. Calculons d'abord la probabilité  $p$  pour qu'un télescope doive être remplacé. Rappelons que, par hypothèse, la durée de vie  $X$  d'un appareil de Celistron obéit à une loi normale  $N(4; 1,5^2)$ . Par conséquent,

$$p = P(X < 2) = P\left(\frac{X-\mu}{\sigma} < \frac{2-4}{1,5}\right) = P(Z < -1,333).$$

Selon les tables de la loi  $N(0; 1)$ ,  $p = 0,0913$ . On peut aussi utiliser la formule suivante d'Excel,

$$=LOI.NORMALE.N(2;MoyenneX;EcartTX;1)$$

qui renvoie la valeur  $p = 0,0912112197258679$ .

Supposons maintenant que Celistron vende  $N$  télescopes. Une proportion  $p = 0,0912$  d'entre eux, soit  $pN$  appareils, auront une durée de vie inférieure à la garantie. Les  $pN$  télescopes de remplacement obéiront à la même loi de probabilité que  $X$  et  $p \times (pN) = p^2N$  d'entre eux tomberont en panne prématurément, ce qui forcera Celistron à fournir gratuitement  $p^2N$  nouveaux appareils. Ces derniers auront eux aussi une probabilité  $p$  de ne pas durer suffisamment longtemps et de devoir être remplacés... En tout, le nombre attendu total  $t$  de télescopes requis pour honorer la garantie s'élève à

$$t = pN + p^2N + p^3N + \dots = \frac{p}{1-p}N = \frac{0,0912}{1-0,0912}N = 0,1004N.$$

Le coefficient 0,1004 représente le nombre attendu de télescopes de remplacement par appareil vendu. Ce paramètre est estimé par le quotient *RemplT/Ventes*. Dans la figure A, le quotient  $84 / 1000 = 0,0084$  diffère passablement de la valeur théorique 0,1004. Il s'agit cependant d'un cas exceptionnel et la majorité des estimés reproduits à la figure précédente sont beaucoup plus près de 0,1004 : par exemple, ceux de la colonne B varient de 0,0860 à 0,1240, pour une valeur moyenne 0,1004 qui coïncide (au niveau de précision considéré) avec la valeur théorique.

#### 4. La somme de deux dés

La figure au haut de la page suivante indique comment nous avons procédé. Tout d'abord, il faut simuler le lancer de deux dés : la formule

$$1+6*ALEA()$$

renvoie un nombre pseudo-aléatoire entre 1 et 7, 7 étant exclus; comme la fonction ENT efface la partie fractionnaire, chacun des deux termes de la formule en B3 donne un nombre entier de 1 à 6, les 6 valeurs étant équiprobables. En résumé, la formule en B3 représente la somme des points sur deux dés.

Pour simuler une partie, il suffit de générer les résultats du joueur (colonne B) et du croupier (colonne C), puis de comparer les valeurs obtenues. Dans la formule de la colonne D, 1 correspond à un gain du joueur, et 0 à une perte.

Dans la figure de la page suivante, nous avons simulé 2000 parties. La formule

$$=SOMME(D3:D2002)$$

renvoie le nombre de gains; pour obtenir l'estimé demandé, il suffit maintenant de diviser par 2000. Dans l'échantillon considéré, le joueur gagne 888 fois et la proportion observée de gains est égale à 44,4%.



Fichier : MOG8-04.xlsx		Feuilles : Simul et Fig		
	A	B	C	D
1		Somme des deux dés		Le joueur
2	N°	Croupier	Joueur	gagne?
3	1	7	2	0
4	2	9	11	1
5	3	5	7	1
6	4	8	12	1
7	5	8	4	0
8	6	5	2	0
2000	1998	5	6	1
2001	1999	9	8	0
2002	2000	8	7	0

Cellule	Formule	Copiée dans :
B3	=ENT(1+6*ALEA())+ENT(1+6*ALEA())	B3:C2002
D3	=SI(C3>B3;1;0)	D4:D2002

**Note.** Nous indiquons ici comment calculer la probabilité  $p$  théorique de gain. Notons  $TJ$  (resp.  $TC$ ) la somme des points des deux dés lancés par le joueur (resp. par le croupier). À cause de la symétrie, les probabilités  $P(TJ < TC)$  et  $P(TJ > TC)$  sont égales. Il reste à évaluer  $P(TJ = TC) = \text{Eff}(TJ = TC) / 36^2$ . Le numérateur est la somme des effectifs des événements  $E_n = (TJ = TC = n)$ , où  $n$  varie de 2 à 12. Or, encore en vertu de la symétrie, les événements  $E_{7+i}$  et  $E_{7-i}$  contiennent le même nombre de résultats. Enfin,

$$\text{Eff}(TJ = TC) = 2(1^2 + 2^2 + 3^2 + 4^2 + 5^2) + 6^2 = 146.$$

Ainsi,  $\text{Eff}(TJ = TC) = \frac{1}{2}(36^2 - 146)$  et  $p = (36^2 - 146) / (2 \times 36^2) = 0,4437$ .

## 5. Craps.

La figure 1 (voir page suivante) indique comment nous avons procédé. La formule de B2 est identique à celle qui se trouve dans la cellule B3 de la figure du problème précédent; elle donne la somme des points lors du lancer de deux dés.

À la suite du 1<sup>er</sup> lancer, trois résultats peuvent survenir :

- le joueur gagne (1 sera affiché en C2) lorsque le résultat en B2 est égal à 7 ou 11;
- le joueur perd (0 sera affiché en C2) lorsque le résultat en B2 est égal à 2, 3 ou 12;
- sinon, la partie continue et C sera affiché en C2.

La formule en C2 vérifie d'abord si le joueur a gagné. Dans le cas positif, 1 est affiché en C2 et la partie est terminée. Sinon, on teste s'il a perdu. Si oui, 0 est affiché en C2 et la partie est terminée. Sinon, la partie se continue et C est affiché en C2.

Fichier : MOG8-05.xlsx				Feuilles : PartieT et Fig1		
	A	B	C	D	E	F
1	Lancer	Résultat	Gain?	Codes pour la colonne C et leur interprétation		
2	1	10	C			
3	2	3	C			
4	3	6	C	Code	Interprétation	
5	4	5	C			
6	5	5	C	0	Le joueur perd la partie.	
7	6	2	C			
8	7	5	C	1	Le joueur gagne la partie.	
9	8	8	C			
10	9	9	C	C	La partie se continue :	
11	10	7	0		le joueur lancera les dés une fois de plus.	
12	11					
13	12					
71	70					

Cellule	Formule	Copiée dans :
B2	=ENT(6*ALEA()+1)+ENT(6*ALEA()+1)	-----
C2	=SI(OU(B2=7;B2=11);1;SI(OU(B2=2;B2=3;B2=11);0;"C"))	-----
B3	=SI(OU(C2=1;C2=0;C2="");"";ENT(6*ALEA()+1)+ENT(6*ALEA()+1))	B4:B71
C3	=SI(OU(C2=1;C2=0;C2="");"";SI(B3=B\$2;1;SI(B3=7;0;"C")))	C4:C71

Lors des lancers subséquents, il faut d'abord vérifier si la partie est encore en cours. C'est pourquoi, à partir de la ligne 3, les formules des colonnes B et C débutent par un test sur la valeur reportée dans la cellule de la colonne C de la ligne précédente. Considérons, à titre d'exemple, la formule en B5, qui s'écrit :

$$=SI(OU(C4=1;C4=0;C4="");"";ENT(6*ALEA()+1)+ENT(6*ALEA()+1))$$

Si C4 contient la valeur 1 ou 0, la partie s'est terminée lors du dernier lancer et on laissera les cellules B5 et C5 vides, pour indiquer que le joueur n'a plus à lancer les dés. De même, si C4 est vide, la partie est terminée depuis quelque temps et, dans ce cas également, on laissera les cellules B5 et C5 vides. Sinon, la partie est encore en cours et le joueur doit lancer les dés une fois de plus. La cellule B5 contiendra la somme des points sur les deux dés; et C5, la conclusion de ce lancer, soit 1 (gagne quand B5 coïncide avec le résultat en B2), 0 (perd si le joueur a obtenu un total de 7) ou C (sinon, la partie se continue).

Nous avons réservé un maximum de 70 lignes pour noter les résultats des différents lancers. La probabilité qu'il faille plus de 70 lancers dans une partie donnée est extrêmement faible. D'ailleurs, nous n'avons observé aucune partie qui exigeait plus de 40 lancers; le maximum de lancers pour une simulation de 10 000 parties se situe le plus souvent entre 25 et 30.

La figure 2 (voir page suivante) résume les résultats de 10 000 parties. La colonne B indique si le joueur a gagné ou perdu; la colonne C donne le nombre de fois qu'il a lancé les dés.

Fichier : MOG8-05.xlsx		Feuilles : Simul et Fig2	
	A	B	C
1		Le joueur	Nombre
2	No	gagne?	de lancers
3		1	1
4	1	1	5
5	2	1	19
6	3	0	2
7	4	0	2
8	5	1	2
9	6	0	3
10	7	1	5
10001	9998	0	4
10002	9999	0	1
10003	10000	0	4
Cellule	Formule		
B3	=MAX(PartieT!C2:C71)		
C3	=NB.SI(PartieT!C2:C71;"C")+1		

Dans l'échantillon de 10 000 parties reproduit à la figure ci-dessus, le joueur a gagné 4954 fois. Les feuilles Résultats et Fig3 du chiffrier MOG8-05.xlsx décrivent la distribution du nombre  $L$  de lancers pour les 10 000 parties d'une simulation. Dans le cas traité à la figure 2, il faut en moyenne 3,49 lancers; et 95% des parties en exigent 9 ou moins.

## 6. La grippe aviaire.

La première étape consiste à simuler une commande de 25 millions de doses. La figure ci-dessous (voir page suivante) illustre une façon de procéder. Les paramètres utilisés sont :

- $ValeursD$  et  $CumulD$  sont des plages qui contiennent, la première, la liste des valeurs de la variable aléatoire  $D$ , «durée (en jours) de fabrication d'un lot», et la seconde, la suite des probabilités que  $D$  soit inférieure aux différentes valeurs de  $D$ ; par exemple,
 
$$(3^{\text{e}} \text{ élément de } CumulD) = P(D < 8) = P(D = 6) + P(D = 7) = 0,04 + 0,08 = 0,12.$$
- Les noms  $NMin$  et  $NMax$  renvoient à des cellules de la feuille Données qui contiennent les valeurs minimale 800 000 et maximale 1 500 000 de la variable aléatoire  $N$ , «nombre de doses fournies par un lot»;  $Nk = (NMod - NMin) / (NMax - NMin)$  représente la proportion de l'intervalle  $[NMin; NMax]$  située à la gauche de la valeur modale  $NMod = 1\,000\,000$ .
- $ProbR = 0,1$  est la probabilité qu'un lot soit rejeté lors de l'inspection.
- $TComm = 25\,000\,000$  est la taille de la commande.

Chaque ligne de la figure 1 correspond à un lot. Analysons plus en détail les données du premier (ligne 3). La formule en B3 est analogue à celle apparaissant dans la cellule D4 de la figure 8.5 du manuel (voir page 418) : elle commande à Excel de générer un nombre pseudo-aléatoire entre 0 et 1, puis de le transformer en une valeur de la variable  $D$ . Le résultat représente la durée

aléatoire de fabrication du lot. De même, C3 et D3 génèrent une valeur pseudo-aléatoire de la variable  $N$  qui, rappelons-le, obéit à une loi triangulaire; la formule en D3 est une adaptation au présent contexte de celle utilisée en section 8.2.4 (voir figure 8.19, colonne G). Enfin, F3 détermine, avec une probabilité  $\text{ProbR} = 10\%$ , si le lot sera rejeté lors de l'inspection : dans le cas positif, on reportera en G3 le nombre de doses calculé en D3; sinon, le stock cumulé sera nul. La dernière colonne gère une horloge, qui indique combien de jours se sont écoulés depuis le début des travaux du laboratoire sur la commande.

Fichier : MOG8-06.xlsx				Feuilles : CommandeT et Fig1			
	A	B	C	D	E	F	G
1	Lot	Temps D	Nombre	Nombre	Inspection	Stock	Horloge
2	No	de fabrication	ps-aléatoire	de doses		cumulé	(en jours)
3	1	14	0,544	1 100 372	1	1 100 372	14
4	2	12	0,070	899 286	0	1 100 372	26
5	3	7	0,362	1 027 360	1	2 127 732	33
6	4	11	0,916	1 328 475	1	3 456 207	44
7	5	10	0,618	1 134 206	1	4 590 413	54
8	6	13	0,055	887 555	1	5 477 968	67
9	7	6	0,038	873 213	1	6 351 181	73
10	8	14	0,437	1 055 902	1	7 407 083	87
11	9	10	0,722	1 188 001	1	8 595 084	97
12	10	11	0,068	897 406	1	9 492 490	108
13	11	6	0,901	1 314 246	1	10 806 736	114
14	12	11	0,882	1 296 432	1	12 103 168	125
15	13	11	0,147	943 682	1	13 046 850	136
16	14	11	0,449	1 060 766	1	14 107 616	147
26	24	9	0,530	1 094 470	1	24 556 548	251
27	25	8	0,866	1 283 791	1	25 840 339	259
28	26						

Cellule	Formule
B3	=RECHERCHE(ALEA();CumulD;ValeursD)
C3	=ALEA()
D3	=ENT(SI(C3<=Nk;NMin+(NMax-NMin)*RACINE(Nk*C3);NMax-(NMax-NMin)*RACINE((1-Nk)*(1-C3))))
E3	=SI(ALEA()<ProbR;0;1)
F3	=SI(E3=1;D3;0)
G3	=B3
B4	=SI(OU(F3>=TComm;F3=" ");" ";RECHERCHE(ALEA();CumulD;ValeursD))
G4	=SI(OU(F3>=TComm;F3=" ");" ";G3+B4)

Les formules de la ligne 3 sont copiées – avec un ajustement dont nous parlons ci-dessous – dans les lignes subséquentes. Nous avons prévu un maximum de 50 lignes ou lots pour une commande de 25 millions de doses. Il est fort improbable que ce maximum soit excédé et le risque que sa présence dans notre modèle nous induise en erreur est minime. L'ajustement mentionné ci-dessus tient au fait qu'il faut tenir compte de la possibilité que la commande soit déjà complétée; dans ce cas, on voudra ne rien afficher dans la ligne correspondante. Considérons, à titre d'exemple, la formule qui a été reportée en B7 :

=SI(OU(F6>=TComm;F6=" ");" ";RECHERCHE(ALEA();CumulD;ValeursD)).

Si les 25 millions de doses requises ont été obtenues avec le lot précédent ( $F6 \geq TComm$ ), il est inutile de fabriquer un lot additionnel et les travaux du laboratoire s'arrêtent : nous convenons de laisser vides les colonnes B à G de la ligne courante (ligne 7). De même, si la condition «  $F6 = ""$  » est satisfaite, c'est que la commande a été complétée antérieurement et, de nouveau, nous voulons ne rien afficher sur la ligne courante. Enfin, dans tous les autres cas, les travaux du laboratoire doivent se poursuivre et Excel renverra en B7 le temps de fabrication d'un nouveau lot.

La 2<sup>e</sup> étape vise à simuler 1000 commandes. Concrètement, nous avons reporté la formule

$$=MAX(CommandeT!G3:G52)$$

dans la cellule B2 de la feuille Simul; puis, nous avons sélectionné la plage A2:B1002 et avons utilisé la commande Table de données... d'Excel.

La figure 2 résume les résultats obtenus. Selon notre échantillon, la Ministre devra tenir compte du fait que le temps requis pour une commande s'échelonne de 196 à 352 jours. De plus, le délai moyen est de 257,5 jours, avec un écart type de 21,5 jours.

Fichier : MOG8-06.xlsx		Feuilles : Résultats et Fig2	
	A	B	
1		Délai	
2	Minimum	196	
3	5 <sup>e</sup> centile	225	
4	Médiane	256	
5	95 <sup>e</sup> centile	295	
6	Maximum	352	
7	Moyenne $\mu$	257,5	
8	Écart type	21,5	
9	Intervalle à 90% pour $\mu$ : B.Inf	256,4	
10	Intervalle à 90% pour $\mu$ : B.Sup	258,6	

## 7. Les tapis roulants.

La figure 1 (voir page suivante) indique comment nous avons procédé. Les formules de la plage B3:G3 sont copiées dans B4:G1002. Les paramètres utilisés sont :

- Les noms  $DMin$  et  $DMax$  renvoient à des cellules de la feuille Données qui contiennent les valeurs minimale 2 000 et maximale 6 000 de la variable aléatoire  $D$ , «demande (en u) durant la période ciblée de deux mois»;  $Dk = (DMod - DMin) / (DMax - DMin)$  représente la proportion de l'intervalle  $[DMin; DMax]$  située à la gauche de la valeur modale  $DMod = 3\,500$ .
- $PrMin = 500$  et  $PrMax = 700$  sont les valeurs minimale et maximale de la variable aléatoire  $Pr$ , «prix de vente (en \$) durant la période».
- $QtéF = 5\,000$  est le nombre d'unités qui seront fabriquées.
- $CoûtU = 345$  est le coût de revient (en \$/u).

Fichier : MOG8-07.xlsx			Feuilles : Simul et Fig1				
	A	B	C	D	E	F	G
1		Période initiale de 2 mois				Quantité	
2	N°	NPA	Demande	Ventes	PrixV	soldée	Profit
3	1	0,793	4 561	4 561	503,69 \$	439	682 892 \$
4	2	0,032	2 440	2 440	669,72 \$	2 560	766 358 \$
5	3	0,133	2 891	2 891	552,30 \$	2 109	454 100 \$
6	4	0,344	3 437	3 437	678,63 \$	1 563	1 137 808 \$
7	5	0,972	5 466	5 000	562,94 \$	0	1 089 700 \$
8	6	0,431	3 613	3 613	691,89 \$	1 387	1 254 631 \$
9	7	0,565	3 913	3 913	504,09 \$	1 087	521 483 \$
1001	999	0,720	4 326	4 326	638,20 \$	674	1 250 927 \$
1002	1000	0,127	2 872	2 872	679,28 \$	2 128	948 646 \$

Cellule	Formule
B3	=ALEA()
C3	=ARRONDI(SI(B3<=Dk;DMin+(DMax-DMin)*RACINE(Dk*B3);DMax-(DMax-DMin)*RACINE((1-Dk)*(1-B3)));0)
D3	=MIN(C3;QtéF)
E3	=ARRONDI(PrMin+ALEA()*(PrMax-PrMin);2)
F3	=QtéF-D3
G3	=(D3*E3)+(F3*ARRONDI(E3/2;2))-QtéF*CoûtU

Les formules de B3 et C3 génèrent une valeur pseudo-aléatoire de la variable  $D$  qui, rappelons-le, obéit à une loi triangulaire; la formule en C3 est une adaptation au présent contexte de celle utilisée en section 8.2.4 (voir figure 8.19, colonne G); enfin, le recours à la fonction ARRONDI d'Excel découle du fait que la demande  $D$  est nécessairement un nombre entier. Les ventes ne peuvent évidemment dépasser ni la demande, ni la quantité fabriquée, ce qui justifie de définir en colonne D les ventes comme le minimum de la valeur en colonne C et de QtéF. La formule de la colonne E génère une valeur pseudo-aléatoire de la variable  $Pr$  qui obéit à une loi uniforme; le prix obtenu est arrondi à 2 décimales, c'est-à-dire à la cent près. La colonne F contient le nombre d'unités soldées, qui tout naturellement se définit comme la différence entre le nombre d'unités fabriquées et le nombre d'unités vendues. Enfin, le profit qu'apportera à notre fabricant la vente du lot de 5 000 tapis est calculé en colonne G comme la différence entre les revenus et les coûts :

- Le coût total encouru par le fabricant s'obtient comme le produit du nombre d'unités fabriquées, QtéF = 5 000, et du coût unitaire de revient, CoûtU.
- Le revenu provenant des tapis vendus au prix régulier est égal au produit de la demande (colonne D) et du prix unitaire (colonne E).
- On procède de façon analogue pour les tapis soldés à moitié prix : le nombre d'unités se trouve cette fois en colonne F, tandis que le prix unitaire se calcule en divisant par 2 le prix régulier (colonne E), puis en arrondissant à la cent près.

La figure 2 (voir page suivante) résume les résultats obtenus. Selon notre échantillon, le revenu pourrait s'établir entre un minimum de 125 799 \$ et un maximum de 1 769 300 \$; par ailleurs, il y a 50% des chances pour que le lot de 50 000 tapis rapporte plus de 909 517 \$ au fabricant, et 50% des chances pour qu'il en rapporte moins. Enfin, le revenu espéré s'élève à 922 793 \$.

Fichier : MOG8-07.xlsx						Feuilles : Résultats et Fig2	
	A	B	C	D	E	F	G
1		Période initiale de 2 mois				Quantité	
2	N°	NPA	Demande	Ventes	PrixV	soldée	Profit
3	1	0,793	4 561	4 561	503,69 \$	439	682 892 \$
4	2	0,032	2 440	2 440	669,72 \$	2 560	766 358 \$
5	3	0,133	2 891	2 891	552,30 \$	2 109	454 100 \$
6	4	0,344	3 437	3 437	678,63 \$	1 563	1 137 808 \$
7	5	0,972	5 466	5 000	562,94 \$	0	1 089 700 \$
8	6	0,431	3 613	3 613	691,89 \$	1 387	1 254 631 \$
9	7	0,565	3 913	3 913	504,09 \$	1 087	521 483 \$
1001	999	0,720	4 326	4 326	638,20 \$	674	1 250 927 \$
1002	1000	0,127	2 872	2 872	679,28 \$	2 128	948 646 \$

## 8. Un appel d'offres.

La figure 1 ci-dessous indique comment nous avons procédé. Les paramètres utilisés sont :

- $CoûtP = 12\ 100$  est le coût estimé du projet (en k\$).
- Les noms  $CMin$  et  $CMax$  renvoient à des cellules de la feuille Données qui contiennent les valeurs minimale  $0,9 \times CoûtP = 10\ 890$  et maximale  $2,5 \times CoûtP = 30\ 250$  de la variable aléatoire  $C$ , «montant (en k\$) de la soumission»;  $Ck = (CMod - CMin) / (CMax - CMin)$  représente la proportion de l'intervalle  $[CMin; CMax]$  située à la gauche de la valeur modale  $CMod = 1,3 \times CoûtP = 15\ 730$ .
- $SoumP$  représente le montant (en k\$) de la soumission de Panzetta; ce paramètre, qui se trouve dans la cellule A3 de la feuille Résultats (voir figure 2), sert à relier le tableau de cette feuille aux calculs de la feuille Simul.

Fichier : MOG8-08.xlsx						Feuilles : Simul et Fig1					
	A	B	C	D	E	F	G	H	I	J	K
1		A		B		C		D		Panzetta	
2	N°	NPA	Montant	NPA	Montant	NPA	Montant	NPA	Montant	Min	gagne?
3	1	0,130	14 385	0,242	15 656	0,861	23 999	0,458	17 911	14 385	1
4	2	0,429	17 580	0,197	15 192	0,617	19 880	0,374	16 980	15 192	1
5	3	0,727	21 489	0,802	22 798	0,007	11 673	0,360	16 832	11 673	0
5001	4999	0,859	23 946	0,792	22 610	0,400	17 262	0,280	16 022	16 022	1
5002	5000	0,905	25 093	0,122	14 265	0,312	16 344	0,204	15 267	14 265	1
Cellule	Formule										Copiée dans :
B3	=ALEA()										B4:B5002
J3	=MIN(C3;E3;G3;I3)										J4:J5002
K3	=SI(SoumP<=J3;1;0)										K4:K5002

Les colonnes B à I génèrent des valeurs pseudo-aléatoires pour les soumissions des quatre concurrents de Panzetta. En pratique, nous avons d'abord reporté en B3 et en C3 les formules pertinentes, puis nous avons copié la plage B3:C3 dans B3:I5002. La cellule C3 transforme le nombre pseudo-aléatoire de B3 en une valeur de la variable  $C$ , qui, rappelons-le, obéit à une loi triangulaire. La formule de C3, qui se lit :

$$=SI(B3<=Ck;CMin+(CMax-CMin)*RACINE(Ck*B3);CMax-(CMax-CMin)*RACINE((1-Ck)*(1-B3)))$$

est une adaptation au présent contexte de celle utilisée en section 8.2.4 du manuel. Noter que, par exemple, la formule de E3 réfère à la cellule D3, et non à B3; en effet, les concurrents doivent préparer leur soumission de façon indépendante, sinon ils pourraient être accusés de collusion.

Pour déterminer si Panzetta obtient le contrat, il reste maintenant à vérifier si sa soumission est la plus basse. C'est l'objet des colonnes J et K : il suffit de comparer le montant *SoumP* exigé par Panzetta dans son offre au maximum des montants apparaissant dans les colonnes C, E, G et I (les cas d'égalité, qui devraient être rarissimes, sont tranchés ici en faveur de Panzetta). La formule de la colonne K renvoie 1 quand Panzetta obtient le contrat, et 0 sinon.

La comparaison des différentes options qui s'offrent à Panzetta se fait dans les feuilles Résultats et Fig2, dont des extraits sont reproduits dans la figure 2 (voir page suivante).

Tout d'abord, nous avons reporté, dans la plage A3:C3 de la feuille Résultats, la valeur du paramètre *SoumP* utilisé dans la feuille Simul (A3), la proportion des 5000 cas de la simulation où Panzetta décroche le contrat (B3) et le revenu net moyen de Panzetta pour ces 5000 cas (C3). Puis, nous avons sélectionné la plage A3:C16 et utilisé la commande Table de données... d'Excel.

La meilleure stratégie est évidemment celle qui correspond au revenu le plus élevé. Pour mettre la réponse en évidence, nous avons calculé en C17 la valeur maximale de la plage C4:C16, puis nous avons comparé le contenu de C17 avec les différentes valeurs apparaissant dans cette plage. Selon l'échantillon considéré aux figures 1 et 2, Panzetta aurait intérêt à fixer sa soumission à 14 750 000 dollars.



Fichier : MOG8-08.xlsx		Feuilles : Résultats et Fig2		
	A	B	C	D
1	Soumission	Panzetta	Revenu net	Stratégie
2	de Panzetta	gagne?	espéré	optimale
3		0,645	1 224,7	
4	14 000	0,639	1 214,1	14 750
5	14 250	0,594	1 277,5	
6	14 500	0,554	1 328,6	
7	14 750	0,513	1 360,0	
8	15 000	0,447	1 296,3	
9	15 250	0,410	1 290,9	
10	15 500	0,371	1 260,7	
11	15 750	0,335	1 222,0	
12	16 000	0,266	1 038,2	
13	16 250	0,232	962,0	
14	16 500	0,209	921,4	
15	16 750	0,183	850,0	
16	17 000	0,149	729,1	
17	Maximum		1 360,0	
Cellule	Formule		Copiée dans :	
B3	=MOYENNE(Simul!K3:K5002)		-----	
C3	=(A3-CoûtP)*B3		-----	
C17	=MAX(C4:C16)		-----	
D4	=SI(C4=C\$17;A4;"")		D5:D16	

## 9. Les soles du restaurant Les Deux Gauloises

Dans un premier temps, nous simulons 1000 journées. Les paramètres utilisés sont :

- *ValeursD* et *CumulD* sont des plages qui contiennent, la première, la liste des valeurs de la variable aléatoire *D*, «demande», et la seconde, la suite des probabilités que *D* soit inférieure aux différentes valeurs de *D*; par exemple,
 
$$(3^{\text{e}} \text{ élément de } \textit{CumulD}) = P(D < 60) = P(D = 40) + P(D = 50) = 0,10 + 0,15 = 0,25.$$
- *CtP*, qui représente le coût de pénurie (en \$), sera fixé à 0 dans la question (a), mais variera de 0,50 \$ à 2 \$ par pas de 0,25 \$ en question (b).

La figure ci-dessous (voir page suivante), qui reproduit des extraits des feuilles Simul et FigS, illustre comment calculer les revenus nets de 1000 jours quand le coût de pénurie *CtP* est fixé à 2 \$. Noter que, pour obtenir les formules des colonnes D à M, il suffit de saisir celles des cellules D3 et E3, puis de copier la plage D3:E3 dans D3:M1002.

Fichier : MOG8-09.xlsx				Feuilles : Simul et FigS									
	A	B	C	D	E	F	G	H	I	J	K	L	M
1	N°	NPA	Demande	TComm = 40	TComm = 50	TComm = 60	TComm = 70	TComm = 80					
2				Ventes	Rev. net	Ventes	Rev. net	Ventes	Rev. net	Ventes	Rev. net	Ventes	Rev. net
3	1	0,547	70	40	340 \$	50	460 \$	60	580 \$	70	700 \$	70	660 \$
4	2	0,514	70	40	340 \$	50	460 \$	60	580 \$	70	700 \$	70	660 \$
5	3	0,928	80	40	320 \$	50	440 \$	60	560 \$	70	680 \$	80	800 \$
6	4	0,165	50	40	380 \$	50	500 \$	50	460 \$	50	420 \$	50	380 \$
7	5	0,412	60	40	360 \$	50	480 \$	60	600 \$	60	560 \$	60	520 \$
1001	999	0,768	70	40	340 \$	50	460 \$	60	580 \$	70	700 \$	70	660 \$
1002	1000	0,029	40	40	400 \$	40	360 \$	40	320 \$	40	280 \$	40	240 \$
Cellule	Formule										Copiée dans :		
B3	=ALEA()										B4:B1002		
C3	=RECHERCHE(B3;CumulD;ValeursD)										C4:C1002		
D3	=MIN(\$C3;E\$1)										colonnes D, F, H, J et L		
E3	=20*D3+6*(E\$1-D3)-10*E\$1-CtPén*(C3-D3)										colonnes E, G, I, K et M		

La colonne C transforme les nombres pseudo-aléatoires de la colonne B en valeurs de la variable  $D$ , tandis que les colonnes D à M donnent les ventes et le revenu net d'une journée selon les cinq scénarios envisagés par le restaurateur. Évidemment, le chiffre des ventes (colonne D) est limité à la fois par le demande (colonne C) et par le nombre de soles commandées (cellules E1, G1, I1, K1 et M1); il est donc raisonnable de reporter en D3 la formule =MIN(\$C3;E\$1). Quant au revenu net, il se calcule comme la différence entre les revenus et les coûts. Considérons de plus près la 1<sup>re</sup> journée de l'échantillon de la figure (ligne 3) et commentons brièvement les calculs des cellules E3 et M3. Le demande (C3) sera de 70 soles. Selon le 1<sup>er</sup> scénario, les achats (E1) sont de 40 soles et le revenu net (E3) s'élèvera à 340 \$ :

- Le propriétaire aura déboursé  $(10 \times 40) = 400$  dollars pour l'achat des 40 poissons.
- Entre le jeudi et le samedi, le restaurant vendra 40 soles, pour un revenu de  $(20 \times 40) = 800$  dollars.
- 30 des 70 clients commandant une sole n'obtiendront pas le plat choisi et le restaurateur encourra un coût de pénurie de  $(2 \times 30) = 60$  dollars.
- En résumé,  $RevN = 800 - 400 - 60 = 340$ .

De même, selon le 5<sup>e</sup> scénario, les achats (M1) sont de 80 soles et le revenu net (M3) s'élèvera à 660 \$ :

- Le propriétaire aura déboursé  $(10 \times 80) = 800$  dollars pour l'achat des 80 poissons.
- Entre le jeudi et le samedi, le restaurant vendra 70 soles, pour un revenu de  $(20 \times 70) = 1400$  dollars.
- Les 10 poissons en excédent entreront dans les plats cuisinés et rapporteront  $(6 \times 10) = 60$  dollars.
- En résumé,  $RevN = 1400 - 800 + 60 = 660$ .

(a) Le tableau suivant donne le revenu net moyen pour un échantillon de 1000 journées en fonction du nombre  $TComm$  de soles commandées.

$TComm$	40	50	60	70	80
RevN moyen	400,0	488,0	551,0	581,0	566,6

On constate que, pour cet échantillon, la meilleure stratégie pour le restaurateur est de fixer à 70 soles sa commande hebdomadaire.

**Note.** Le revenu net, quand  $TComm = 40$ , se calcule comme suit :

$$RevN = 40 \times (20 - 10) - (CtPén \times D) = 400 - (CtPén \times D),$$

où  $D$  est la demande. Si  $CtPén = 0$ , le revenu net est une constante.

(b) La figure ci-dessous indique comment nous avons procédé pour analyser l'impact du coût de pénurie sur le revenu moyen. Nous avons d'abord reporté la formule

$$=MOYENNE(Simul!E3:E1002)$$

dans la cellule B8 et l'avons copiée dans C8:F8; puis, nous avons complété les plages A9:A17 et B7:F7. Ensuite, nous avons sélectionné la plage A2:B1002 et utilisé la commande Table de données... d'Excel. Enfin, les colonnes G et H déterminent la stratégie optimale, compte tenu du coût de pénurie retenu par le restaurateur.

Fichier : MOG8-09.xlsx							Feuilles : Qb et FigB	
	A	B	C	D	E	F	G	H
7		40	50	60	70	80	RevN	Stratégie
8	CtPén	352,08	456,40	537,04	581,36	575,44	maximal	optimale
9	0,00	400,00	486,98	550,72	578,76	567,74	578,76	70
10	0,25	394,15	482,40	548,56	580,95	567,60	580,95	70
11	0,50	388,03	477,52	547,72	583,84	575,16	583,84	70
12	0,75	382,38	477,49	550,77	583,64	568,86	583,64	70
13	1,00	376,10	470,65	546,30	582,35	574,60	582,35	70
14	1,25	371,20	468,30	541,30	575,57	562,56	575,57	70
15	1,50	364,36	466,81	545,38	583,04	572,64	583,04	70
16	1,75	358,70	460,14	539,52	576,38	570,40	576,38	70
17	2,00	352,42	457,86	540,10	581,54	573,06	581,54	70
Cellule	Formule						Copiée dans :	
B8	=MOYENNE(Simul!E3:E1002)						C8:F8	
G9	=MAX(B9:F9)						G10:G17	
H9	=SI(G9=B9;B\$7;SI(G9=C9;C\$7;SI(G9=D9;D\$7;SI(G9=E9;E\$7;F\$7))))						H10:H17	

La meilleure stratégie, selon notre échantillon, consiste à commander 70 soles par semaine, quelle que soit la valeur du coût de pénurie.

## 10. Le marketing des petits pois.

La première étape de notre analyse consiste à simuler une période typique de 40 trimestres consécutifs. La figure 1 ci-dessous décrit une façon de procéder dans le cas où le producteur P2 a mené une campagne normale durant le dernier trimestre.

Fichier : MOG8-10.xlsx							Feuilles : PériodeT et Fig1			
	A	B	C	D	E	F	G	H	I	J
1	Trimestre	Stratégie		Loi triangulaire de $X_t$				Nombre	Changement	Part de P1
2	t	de P1	de P2	Minimum	Mode	Maximum	k	ps-aléatoire	part de P1	à la fin de t
3	0		0							0,5600
4	1	0	0	-0,04	0	0,04	0,500	0,305	-0,0088	0,5512
5	2	0	1	-0,06	-0,02	0,01	0,571	0,561	-0,0204	0,5309
6	3	1	0	-0,01	0,02	0,06	0,429	0,468	0,0214	0,5523
7	4	1	0	-0,01	0,02	0,06	0,429	0,572	0,0254	0,5777
11	8	0	1	-0,06	-0,02	0,01	0,571	0,565	-0,0202	0,5059
43	40	1	0	-0,01	0,02	0,06	0,429	0,260	0,0134	0,6771

Cellule	Formule	Copiée dans :
B4	=SI(C3=1;SI(ALEA())<=0,7;1;0);SI(ALEA())<=0,6;1;0))	B5:B43
C4	=SI(C3=1;SI(ALEA())<=0,4;1;0);SI(ALEA())<=0,5;1;0))	C5:C43
D4	=INDEX(Tr;2*\$B4+\$C4+1;1)	D5:D43
E4	=INDEX(Tr;2*\$B4+\$C4+1;2)	E5:E43
F4	=INDEX(Tr;2*\$B4+\$C4+1;3)	F5:F43
G4	=INDEX(Tr;2*\$B4+\$C4+1;4)	G5:G43
H4	=ALEA()	H5:H43
I4	=SI(H4<=G4;D4+(F4-D4)*RACINE(G4*H4);F4-(F4-D4)*RACINE((1-G4)*(1-H4)))	I5:I43
J4	=MAX(MIN(J3+I4;1);0)	J5:J43

Dans cette figure, on a associé une ligne à chacun des trimestres, la ligne 3 ( $t = 0$ ) correspondant au trimestre précédant immédiatement la période considérée.

- Les colonnes B et C donnent la stratégie publicitaire des deux concurrents pour un trimestre donné, 1 représentant une campagne intense et 0, une campagne normale. Ainsi, la valeur 0 reportée en C3 indique que, dans la feuille Fig1, nous prenons comme hypothèse que P2 a mené une campagne publicitaire normale lors du trimestre  $t = 0$ . Analysons plus en détail la formule de B4:

$$=SI(C3=1;SI(ALEA())<=0,7;1;0);SI(ALEA())<=0,6;1;0))$$

D'après l'énoncé du problème, P1 choisit une campagne intense avec une probabilité de 70% quand la campagne de P2 fut intense le trimestre précédent, et avec une probabilité de 60% dans le cas contraire. Conformément à cette règle, en B4, on teste d'abord la décision de P2 lors du trimestre 0; dans le cas d'une campagne intense (valeur 1 en B3), Excel renverra la valeur 1 avec une probabilité de 70%; sinon, la valeur 1 sera obtenue avec une probabilité de 60%. De même, la colonne C traduit la stratégie de P2 décrite dans l'énoncé.

- Par hypothèse, la variable  $X_t$ , « changement de la part de marché de P1 lors du trimestre  $t$  », obéit à une loi triangulaire dont les paramètres dépendent du type de campagne publicitaire menée par les deux concurrents. Des valeurs pseudo-aléatoires de  $X_t$  sont obtenues de la façon suivante : on génère d'abord, en colonne H, un nombre pseudo-aléatoire entre 0 et 1; puis, en colonne I, ce dernier est transformé en une valeur d'une loi triangulaire à l'aide d'une

version adaptée au présent contexte de la formule apparaissant dans la cellule G6 de la figure 8.19 (voir page 441). Afin d'alléger, nous avons reporté dans les colonnes D à G les paramètres pertinents de la loi triangulaire. Ainsi, dans les formules des cellules D4 à G4 de la figure 1, le nom Tr renvoie à un tableau  $4 \times 4$  de la feuille Données, dont les lignes correspondent aux 4 combinaisons possibles de campagne publicitaire par les deux concurrents; les trois premières colonnes de Tr contiennent les valeurs minimales, modales et maximales indiquées dans le 1<sup>er</sup> tableau de l'énoncé, et la dernière, les valeurs du paramètre  $k$  (voir p. 441, ligne 1; ou encore, voir la cellule B3 de la figure 1 de la solution du problème 6). La fonction  $\text{Index}(\text{Tr};i;j)$  d'Excel renvoie l'entrée du tableau Tr située à l'intersection de la ligne  $i$  et de la colonne  $j$ : en D4 est donc affichée la valeur minimale ( $j = 1$ ) de la ligne numéro  $i = (2 \times 0) + 0 + 1 = 1$ , « Ni P1, ni P2 »; en E4 et F4, on retrouve les valeurs modale et maximale de cette même ligne; enfin, la plage D43:G43 contient les quatre paramètres de la ligne numéro  $i = (2 \times 1) + 0 + 1 = 3$ , « Seulement P1 ».

- Pour obtenir la part de marché  $X_t$  de P1 à la fin du trimestre  $t$ , il suffit d'ajouter le changement survenu durant le trimestre à la part de marché à la fin du trimestre précédent. La partie centrale de la formule de la colonne J traduit cette opération. Cependant, comme une part de marché ne peut ni être négative ni excéder 1, on force la valeur de  $X_t$  à rester entre les limites 0 et 1 à l'aide des fonctions Max et Min d'Excel.

L'énoncé demande de simuler 1000 périodes de 40 trimestres et de compiler les valeurs de  $X_t$  où  $t = 4, 8$  et 40. Il suffit de compléter la plage B3:D3 comme dans la figure 2 ci-dessous, puis de recourir à la commande Table de données... d'Excel.

Fichier : MOG8-10.xlsx		Feuilles : Simul et Fig2		
	A	B	C	D
1		Part de marché de P1 à la fin du trimestre $t$		
2	N <sup>o</sup>	$t = 4$	$t = 8$	$t = 40$
3		0,5777	0,5059	0,6771
4	1	0,5543	0,5793	0,6839
5	2	0,6330	0,6391	0,8224
6	3	0,5721	0,6472	0,7612
1002	999	0,6108	0,7093	0,7167
1003	1000	0,5473	0,6304	0,7532
Cellule	Formule			
B3	=PériodeT!J7			
C3	=PériodeT!J11			
D3	=PériodeT!J43			

La plage B3:D11 de la feuille Fig3 (voir page suivante) résume les statistiques des variables  $X_4$ ,  $X_8$  et  $X_{40}$  dans l'échantillon considéré dans la figure précédente. Pour obtenir les résultats des colonnes E à G, nous avons reporté la valeur 1 dans le cellule de la feuille PériodeT, puis avons recalculé le tableau de la feuille Simul en pressant la touche F9.

Fichier : MOG8-10.xlsx		Feuilles : Résultats et Fig3					
	A	B	C	D	E	F	G
1	Campagne de P2 au trimestre 0	Normale			Intense		
2	Part de marché de P1 à la fin du trimestre $t$	$t = 4$	$t = 8$	$t = 40$	$t = 4$	$t = 8$	$t = 40$
3	Minimum	0,4460	0,3780	0,2692	0,4423	0,4074	0,2238
4	5 <sup>e</sup> centile	0,5022	0,4911	0,5148	0,5021	0,4952	0,4916
5	Médiane	0,5749	0,5935	0,7395	0,5788	0,5974	0,7361
6	95 <sup>e</sup> centile	0,6452	0,6983	0,9633	0,6507	0,6947	0,9587
7	Maximum	0,6930	0,7852	1,0000	0,7094	0,7988	1,0000
8	Moyenne $\mu$	0,5761	0,5949	0,7358	0,5782	0,5961	0,7313
9	Écart type	0,0439	0,0643	0,1346	0,0451	0,0609	0,1383
10	Intervalle à 90% pour $\mu$ : B.Inf	0,5738	0,5916	0,7288	0,5758	0,5929	0,7241
11	Intervalle à 90% pour $\mu$ : B.Sup	0,5783	0,5983	0,7428	0,5805	0,5992	0,7385

Les distributions des variables  $X_4$ ,  $X_8$  et  $X_{40}$  ne diffèrent pas significativement selon que la campagne menée par P2 lors du trimestre 0 fut normale ou intense. La part moyenne de marché de P1 tend à augmenter avec le temps : elle est d'environ 57,7% à la fin du trimestre 4 (voir cellules B8 et E8), mais atteint 59,5% un an plus tard et 73,3% à la fin de la période considérée. Cependant, la variabilité augmente également avec le temps et, à long terme, la part de marché de P1 pourrait, à la limite, tomber aussi bas que 22,4%. Évidemment, les résultats à long terme reposent sur les hypothèses qu'aucun changement structurel ne survienne dans le marché des petits pois et que les deux concurrents conservent leur politique de marketing pendant les 40 prochains trimestres.

## 11. Le quotidien Les Échos.

La figure 1 ci-dessous (voir page suivante) indique comment nous avons procédé. Les paramètres utilisés sont :

- $PrixV = 1,00$  et  $CtAch = 0,45$  sont le prix de vente et le coût d'achat d'un exemplaire.
- $CtPén = 0,20$  représente le coût de pénurie.

On génère, en colonne B, un nombre pseudo-aléatoire entre 0 et 1; puis, en colonne C, ce dernier est transformé en une valeur d'une loi triangulaire à l'aide d'une version adaptée au présent contexte de la formule apparaissant dans la cellule G6 de la figure 8.19 (voir page 441). Plus précisément, on a reporté en C4 la formule suivante, qu'on a ensuite copiée dans la plage C5:C2003 :

$$=ARRONDI(SI(B4<=Xk;XMin+(XMax-XMin)*RACINE(Xk*B4);XMax-(XMax-XMin)*RACINE((1-Xk)*(1-B4)));0)$$

Ici, les noms  $XMin$  et  $XMax$  renvoient à des cellules de la feuille Données qui contiennent les valeurs minimale 140 et maximale 250 de la variable aléatoire  $X$ , «nombre d'exemplaires vendus un jour donné»;  $Xk = (XMod - XMin) / (XMax - XMin)$  représente la proportion de l'intervalle  $[XMin; XMax]$  située à la gauche de la valeur modale  $XMod = 225$ .

Les colonnes D à J donnent le revenu de la journée en fonction du nombre d'exemplaires achetés par le propriétaire du dépanneur. Nous expliquons brièvement la formule reporté en D4 :

- Le revenu de la journée est le produit du prix de vente unitaire,  $PrixV$ , et du nombre  $V$  d'exemplaires vendus; et  $V$  s'obtient comme le minimum de la demande ( $C4$ ) et du nombre d'exemplaires disponibles ( $D3$ ).
- Le propriétaire du dépanneur doit déboursier 0,45 \$ par exemplaire acheté; au total, il versera  $0,45 \times 205 = 92,25$  dollars. De façon à obtenir une formule qui s'adapte correctement lors de la copie dans la plage D4:J2003, on réécrit ce produit sous la forme équivalente «  $CtAch * D\$3$  ».
- Le coût de pénurie total associé à une journée se calcule comme le produit du coût unitaire  $CtPén = 0,20$  et du nombre  $M$  d'exemplaires manquants. Par définition,  $M$  est égal à l'écart entre la demande des clients ( $C4$ ) et le nombre d'exemplaires achetés par le dépanneur ( $D3$ ), en autant que la demande excède l'offre; sinon,  $M$  est nul.

Fichier : MOG8-11.xlsx				Feuilles : Simul et Fig1							
	A	B	C	D	E	F	G	H	I	J	
1	<b>Revenu quotidien (en \$) selon le nombre d'exemplaires achetés.</b>										
2				Nombre d'exemplaires achetés							
3	N°	NPA	Demande	205	210	215	220	225	230	235	
4	1	0,957	239	105,95	109,70	113,45	117,20	120,95	124,70	128,45	
5	2	0,648	218	110,15	113,90	117,65	119,00	116,75	114,50	112,25	
6	3	0,786	226	108,55	112,30	116,05	119,80	123,55	122,50	120,25	
2002	1999	0,317	194	101,75	99,50	97,25	95,00	92,75	90,50	88,25	
2003	2000	0,482	207	112,35	112,50	110,25	108,00	105,75	103,50	101,25	
Cellule		Formule								Copiée dans :	
B4		=ALEA()								B5:B2003	
D4		=PrixV*MIN(D\$3;\$C4)-CtAch*D\$3-CtPén*MAX(0;\$C4-D\$3)								D4:J2003	

Le revenu quotidien moyen (en \$) des 2000 journées de notre échantillon est donné dans la figure 2 en fonction du nombre d'exemplaires achetés. On constate que la meilleure politique consiste à acheter entre 215 et 220 copies par jour. (Il est inutile de chercher à être plus précis, à cause de la variabilité des résultats de la simulation.)

Fichier : MOG8-11.xlsx				Feuilles : Résultats et Fig2					
	A	B	C	D	E	F	G	H	
5	Nombre d'exemplaires	205	210	215	220	225	230	235	
6	Revenu moyen (en \$)	101,24	102,08	102,46	102,37	101,70	100,55	98,93	
Cellule		Formule						Copiée dans :	
B6		=MOYENNE(Simul!D4:D1003)						C6:H6	

## 12. Le juste-à-temps.

(ab) Noter que, si le quai est desservi par 3 équipes et que 3 camions surviennent en début d'heure alors que 2 camions sont déjà en attente, ces deux derniers ainsi qu'un seul des 3 nouveaux arrivés seront déchargés durant l'heure suivante et les 2 autres attendront exactement une heure.

Dans un premier temps, nous simulons une période typique de 100 jours consécutifs, soit 2400 heures d'opération en continu. Les paramètres *Valeurs* et *CumulX* renvoient à des plages qui contiennent, la première, la liste des valeurs de la variable aléatoire  $X$ , «nombre de camions qui arrivent au quai de débarquement durant une heure donnée», et la seconde, la suite des probabilités que  $X$  soit inférieure aux différentes valeurs de  $X$ ; par exemple,

$$(3^{\text{e}} \text{ élément de } CumulX) = P(X < 2) = P(X = 0) + P(X = 1) = 0,05 + 0,10 = 0,15.$$

La colonne B génère une valeur pseudo-aléatoire de la variable  $X$ ; celle-ci est ajoutée en colonnes C et E au nombre de camions en attente à la fin de l'heure précédente. De ce total, on retranche le nombre d'équipes, 3 ou 4 (voir cellules B4 et D4), pour déterminer combien de camions attendront durant l'heure courante.

Fichier : MOG8-12.xlsx		Feuilles : PériodeT et Fig1				
	A	B	C	D	E	F
1		Nombre	3 équipes		4 équipes	
2	Horloge	d'arrivées	NbTotal	NbAttente	NbTotal	NbAttente
3	0			2		2
4	1	3	5	2	5	1
5	2	3	5	2	4	0
6	3	4	6	3	4	0
7	4	5	8	5	5	1
8	5	2	7	4	3	0
9	6	3	7	4	3	0
10	7	4	8	5	4	0
2402	2399	2	2	0	2	0
2403	2400	3	3	0	3	0

Cellule	Formule	Copiée dans :
B4	=RECHERCHE(ALEA();CumulX;Valeurs)	B5:B2403
C4	=D3+\$B4	C5:C2403 et E5:E2403
D4	=MAX(0;C4-3)	D5:D2403
F4	=MAX(0;E4-4)	F5:F2403

La ligne 4 de la figure 2 (voir page suivante) résume les résultats obtenus à la figure 1. En particulier, les cellules C4 et E4 donnent le coût moyen selon qu'il y a 3 ou 4 équipes : il suffit de multiplier le nombre moyen de camions en attente par la pénalité  $Pén = 150$  encourue par la manufacture quand un camion doit attendre 1 heure en file qu'on le décharge. La période typique est ensuite répétée 1000 fois à l'aide de la commande Table de données... d'Excel.



Fichier : MOG8-12.xlsx		Feuilles : SimulP et Fig2			
	A	B	C	D	E
1		3 équipes		4 équipes	
2		NbTotal	Coût	NbTotal	Coût
3	N°	Maximum	(en \$/h)	Maximum	(en \$/h)
4		12	141,31	3	10,13
5	1	7	143,63	3	11,31
6	2	8	117,38	2	7,25
7	3	7	143,69	2	8,44
8	4	9	154,25	2	9,25
9	5	13	161,25	3	10,44
1003	999	8	128,63	2	8,81
1004	1000	13	150,06	3	8,94
Cellule	Formule				Copiée dans :
B4	=MAX(PériodeT!D3:D2403)				D4
C4	=Pén*MOYENNE(PériodeT!D4:D2403)				E4

La figure 3 donne les principales statistiques des deux variables « Coût » considérées à la figure précédente. Dans notre échantillon de 1000 périodes typiques, le coût encouru par la manufacture s'élève en moyenne à 151 \$/h quand il y a 3 équipes, et à seulement 9,71 \$/h quand il y en a 4. La direction ajoutera ou non une 4<sup>e</sup> équipe en tenant compte à la fois du coût de cette équipe additionnelle et des inconvénients intangibles de la situation actuelle où de nombreux camions doivent attendre d'être déchargés, parfois aussi longtemps que 5 heures : en effet, tel qu'indiqué par exemple dans la cellule B9 de la figure 2, la file peut atteindre jusqu'à 13 véhicules à un moment donné ; les 13 attendront d'abord 1 heure, puis 12 seront déchargés par les 3 équipes pendant les 4 heures suivantes, tandis que le dernier devra attendre le début de la 6<sup>e</sup> heure.

Fichier : MOG8-12.xlsx		Feuilles : RésultP et Fig3	
	A	B	C
4		3 équipes	4 équipes
5	Minimum	102,50	6,06
6	5e centile	122,00	8,06
7	Médiane	148,31	9,63
8	95e centile	187,90	11,69
9	Maximum	241,31	13,75
10	Moyenne $\mu$	151,00	9,71
11	Écart type	19,87	1,09
12	Intervalle à 90% pour $\mu$ : B.Inf	149,97	9,66
13	Intervalle à 90% pour $\mu$ : B.Sup	152,03	9,77

(c) La figure 4 (voir page suivante) décrit une journée typique dans ce nouveau contexte. Les calculs sont semblables à ceux de la figure 1, sauf qu'on recourt dans les lignes 4 et 19 aux probabilités des variables  $Y$  et  $D$  respectivement, plutôt qu'à celles de  $X$ .

Fichier : MOG8-12.xlsx		Feuilles : JourT et Fig4				
	A	B	C	D	E	F
1		<b>Nombre</b>	3 équipes		4 équipes	
2	<b>Horloge</b>	<b>d'arrivées</b>	NbTotal	NbAttente	NbTotal	NbAttente
3	0			0		0
4	1	4	4	1	4	0
5	2	2	3	0	2	0
6	3	5	5	2	5	1
7	4	2	4	1	3	0
8	5	2	3	0	2	0
9	6	3	3	0	3	0
10	7	1	1	0	1	0
11	8	3	3	0	3	0
12	9	3	3	0	3	0
13	10	2	2	0	2	0
14	11	5	5	2	5	1
15	12	3	5	2	4	0
16	13	2	4	1	2	0
17	14	2	3	0	2	0
18	15	1	1	0	1	0
19	16	3	3	0	3	0

Cellule	Formule	Copiée dans :
B4	=RECHERCHE(ALEA();CumulY;Valeurs)	-----
B5	=RECHERCHE(ALEA();CumulX;Valeurs)	B6:B19
B19	=RECHERCHE(ALEA();CumulD;Valeurs)	-----
C4	=D3+\$B4	C5:C19 et E5:E19
D4	=MAX(0;C4-3)	D5:D19
F4	=MAX(0;E4-4)	F5:F19

On procède ensuite comme aux figures 2 et 3. On calcule d'abord le coût total de la journée typique, puis on simule 2000 jours (voir figure 5 ci-dessous). Enfin, on calcule les principales statistiques de la variable « *Coût total* » : dans notre échantillon de 2000 jours, la manufacture doit rembourser aux fournisseurs en moyenne 1953 \$/j quand il y a 3 équipes, et seulement 171 \$/j quand il y en a 4.

Fichier : MOG8-12.xlsx		Feuilles : SimulJ et Fig5			
	A	B	C	D	E
1		3 équipes		4 équipes	
2		NbTotal	Coût	NbTotal	Coût
3	<b>N°</b>	<b>Maximum</b>	<b>(en \$/j)</b>	<b>Maximum</b>	<b>(en \$/j)</b>
4		6	3 150	5	150
5	1	9	7 350	6	900
6	2	6	2 850	5	150
2003	1999	7	2 700	5	300
2004	2000	5	1 050	5	150

Cellule	Formule	Copiée dans :
B4	=MAX(JourT!C4:C19)	D4
C4	=Pén*SOMME(JourT!D4:D19)	E4

### 13. Le tir aux pigeons.

Dans un premier temps, nous analysons une séance de tir typique. La figure 1 (voir page suivante) indique comment nous avons procédé. Noter que nous avons masqué les colonnes C, D, F et I afin d'alléger la figure. De plus, les formules des colonnes J et K sont copiées de celles des colonnes G et H respectivement; elles ne sont pas affichées, toujours pour alléger. Les paramètres utilisés dans les formules sont :

- Les noms  $RMin$ ,  $RMod$  et  $RMax$  renvoient à des cellules de la feuille Données qui contiennent les valeurs minimale 5, modale 7 et maximale 12 de la variable  $R$ , « temps requis (en secondes) pour recharger et se mettre en position de tir ».
- De même,  $VMin = 1$ ,  $VMod = 2$  et  $VMax = 5$  représentent les valeurs minimale, modale et maximale de la variable  $V$ , « temps requis (en secondes) pour viser et tirer ».
- Enfin, les valeurs minimale et maximale de la variable  $L$ , « temps (en secondes) entre deux lancers consécutifs de pigeons », sont  $LMin = 7$  et  $LMax = 13$  respectivement.
- $p = 0,8$  est la probabilité d'atteindre un pigeon lors d'un tir.
- $NbP = 20$  est le nombre de pigeons que le tireur veut abattre lors de la séance.

Les lignes 43 à 53 de la figure 1 n'affichent que le numéro de la colonne A, car le tireur a atteint son objectif de 20 pigeons lors du 39<sup>e</sup> lancer (voir cellule M42) et la séance s'est terminée à ce moment-là. Comme le nombre de lancers requis n'est pas fixe mais dépend du succès rencontré par le tireur, nous devons prévoir suffisamment de lignes pour être assurés que la séance sera complétée à la fin du tableau. En pratique, nous en avons réservé 50. La probabilité qu'il faille plus de 50 lancers lors d'une séance est extrêmement faible; d'ailleurs, le maximum observé dans notre simulation de 1000 séances est de 43 lancers (voir figure 3, cellule B7).

Les formules de la figure 1 débutent par la fonction Si et renvoient une cellule vide quand la séance est terminée. Dans les colonnes C à M, on se contente de considérer le contenu de la cellule en colonne B de la même ligne : si celle-ci est vide, la séance a déjà pris fin, et notre conclusion sera de laisser vides également les autres cellules de la ligne. Noter que, dans les formules de G4 et de H4, la référence à B4 se fait indirectement, en vérifiant le contenu de la cellule F4 immédiatement à gauche. Le test de la colonne B est plus complexe et mérite quelques commentaires additionnels. Par exemple, en B5, nous testons si l'objectif de 20 pigeons a été atteint lors du tir précédent ( $M4=NbP$ ) ou lors d'un tir antérieur ( $M4=""$ ); comme aucune de ces conditions n'est satisfaite, la séance se continue et il faut simuler un autre lancer. Par contre, en B43, le test s'écrit  $OU(M42=NbP;M42="")$ ; dans la simulation reproduite à la figure 1, le test donne une réponse positive puisque la ligne précédente contient le nombre 20 en colonne M ( $M42=20$ ) ; par conséquent, la cellule B43 sera vide. De même, en B44, le test consiste à évaluer  $OU(M43=NbP;M43="")$  et, cette fois, c'est la seconde condition qui est satisfaite puisque la cellule M43 est vide. Enfin, le test en I4 sera commenté un peu plus loin ci-dessous.

Nous analysons maintenant les formules des colonnes B à M dans le cas où, la conclusion du test étant négative, il faut poursuivre la séance.

Fichier : MOG8-13.xlsx				Feuilles : SéanceT et Fig1				
	B	E	G	H	J	K	L	M
1		Prêt à	1er tir		2e tir		Fin de	Cumul de
2	Lancer	tirer	Temps	Succès?	Temps	Succès?	l'essai	succès
3							0,0	0
4	0,0	0,0	3,4	0	5,4	0	5,4	0
5	8,3	13,7					13,7	0
6	16,8	16,8	20,5	0	24,6	1	24,6	1
7	24,5	33,5					33,5	1
8	33,2	33,5					33,5	1
9	43,9	43,9	46,9	0	50,3	0	50,3	1
10	54,6	56,7					56,7	1
11	62,8	62,8	64,8	0	67,8	1	67,8	2
12	74,0	74,0	78,0	1			78,0	3
13	85,4	85,6					85,6	3
14	93,2	93,2	97,3	1			97,3	4
15	102,0	103,9					103,9	4
41	368,1	371,4					371,4	19
42	381,0	381,0	383,7	1			383,7	20
43								
44								
53								

Cellule	Formule	Copiée dans :
B5	=SI(OU(M4=NbP;M4="");"";B4+LMin+(LMax-LMin)*ALEA())	B6:B53
C4 *	=SI(B4="";"";ALEA())	C5:C53 *
E5	=SI(B5="";"";MAX(B5;SI(E4>B4;E4;L4+D5)))	E5:E53
F4 *	=SI(OU(B4="";E4>B4);"";ALEA())	F5:F53 *
H4	=SI(F4="";"";SI(ALEA())<=p;1;0)	H5:H53
I4 *	=SI(OU(B4="";E4>B4;H4=1);"";ALEA())	I5:I53 *
L4	=SI(B4="";"";MAX(E4;G4;J4))	L5:L53
M4	=SI(B4="";"";M3+MAX(H4;K4))	M5:M53

\* Cellules masquées

- Les données des colonnes B, E, G, J et L représentent des temps (en minutes) écoulés depuis le début de la séance qui, selon l'approche retenue dans le fichier MOG8-13.xlsx, est le moment où le premier pigeon est lancé.
- La colonne B indique quand s'effectuent les différents lancers. Le premier coïncide toujours avec l'instant  $t = 0$  auquel nous avons convenu de démarrer l'horloge et il n'est pas nécessaire de recourir à une formule en B4. Pour les autres, on additionnera au temps du lancer précédent une valeur pseudo-aléatoire de la variable  $L$ . Comme celle-ci obéit à une loi uniforme, l'ajout requis se traduit par «  $+LMin+(LMax-LMin)*ALEA()$  ».
- En colonne E, nous déterminons quand le tireur est prêt à entamer le processus de viser et de tirer. Trois situations sont traitées par la formule de la colonne E.
  - Considérons d'abord le 9<sup>e</sup> lancer qui est décrit sur la ligne 12 : l'essai précédent s'est terminé en  $t = 67,8$  (voir L11) et le tireur a rechargé son arme en 5,8 secondes (voir D12); en  $t = 67,8 + 5,9 = 73,6$ , il attend le prochain lancer, qui est effectué en

- $t = 74,0$  (voir B12). Il commence alors à viser. Ainsi, c'est la valeur de B12 qui est renvoyée par la formule en E12.
- Le 6<sup>e</sup> essai se termine en  $t = 50,3$  (voir L9). Le tireur prend 6,4 secondes pour recharger (voir D10); il est donc prêt en  $t = 50,3 + 6,4 = 56,7$ . Mais le 7<sup>e</sup> pigeon a déjà été lancé (en  $t = 54,6$ : voir B10). Il ne tirera donc pas lors de cet essai. La valeur 56,7 affichée en E10 provient donc de l'expression « L9+D10 ».
  - Il arrive parfois que deux pigeons soient lancés pendant que le tireur recharge son arme. Cette situation survient plutôt rarement – 52 fois dans l'échantillon de 1000 séances considéré dans les figures 2 et 3. Le 3<sup>e</sup> essai se termine en  $t = 24,6$  (voir L6), le tireur prend 8,9 secondes pour recharger (voir D7) et est prêt à viser en  $t = 33,5$  (voir E7); mais le 4<sup>e</sup> pigeon est lancé en  $t = 24,5$  et le 5<sup>e</sup>, en  $t = 33,2$  (voir B8); la valeur renvoyée en E8 doit donc coïncider avec celle de E7. Dans une telle situation où deux lancers sont effectués pendant que l'arme est rechargée, la valeur en colonne E sur la ligne associée au second lancer est égale à celle dans la même colonne sur la ligne précédente.
- Les colonnes F, G et H décrivent les premiers tirs des différentes séances. Considérons d'abord la formule de F, illustrée dans la figure 1 par sa version pour F4. Évidemment, il n'y a pas de tir si la séance est terminée ( $B4=""$ ) ou si le tireur n'a pas fini de recharger au moment du lancer ( $E4 > B4$ ). Dans ces cas, les cellules des colonnes F, G et H restent vides. Sinon, le tireur vise le pigeon, ce qui se traduit dans notre modèle de simulation par une valeur pseudo-aléatoire de la variable  $V$  qui se calcule selon la formule utilisée dans la figure 8.19 ; le temps  $V$  requis pour viser s'ajoute à la valeur en colonne E, pour donner le moment affiché en colonne G où s'effectue le 1<sup>er</sup> tir.
  - Pour simuler le résultat du 1<sup>er</sup> tir, nous générons un nombre pseudo-aléatoire entre 0 et 1 et le comparons à la probabilité  $p = 0,8$  que le tireur atteigne un pigeon. Si la valeur renvoyée par ALEA() est inférieure ou égale à  $p$ , nous considérons que le tir est réussi et 1 sera affiché.
  - Les colonnes I, J et K décrivent le 2<sup>e</sup> tir et les formules utilisées sont semblables à celles des colonnes F, G et H. D'ailleurs, celles de J4 et de K4 ont été obtenues en copiant le contenu de la plage G4:H4. Il faut cependant s'assurer que le 2<sup>e</sup> tir n'est pas effectué quand le pigeon est atteint lors du 1<sup>er</sup>; c'est pourquoi la condition « H4=1 » a été ajoutée dans la formule de I4.
  - La fin de l'essai numéro 1 survient au plus tardif des trois moments suivants : quand le tireur est prêt (E4), quand il effectue le 1<sup>er</sup> tir (G4) ou le second (J4). D'où le recours à la fonction Max dans la formule de L4.
  - Enfin, le nombre cumulé de succès s'obtient en ajoutant le résultat de l'essai courant au total sur la ligne précédente. Noter que la formule « MAX(H4;K4) » donne 0 quand les cellules H4 et K4 sont toutes deux vides.

Il faut maintenant simuler 1000 séances. Dans la figure 2, nous avons synthétisé en ligne 3 les résultats de la feuille SéanceT, puis nous avons utilisé la commande Table de données... d'Excel. La formule en B3, dont l'objectif annoncé est de compter le nombre de lancers lors d'une séance, indique en fait combien il y a de cellules non vides parmi les 50 de la plage B4:B53 de la feuille

SéanceT. Noter que, dans la séance simulée reproduite à la figure 1, les cellules B43 à B53 sont vides et que cette séance a exigé  $50 - (53 - 42) = 39$  essais. De même, le nombre de tirs effectués s'obtient comme le nombre de cellules non vides au total dans les plages G4:G53 et J4:J53. Enfin, la durée d'une séance est le temps à notre horloge lors de la fin du dernier essai; et ce temps est toujours la valeur maximale apparaissant en colonne L.

Fichier : MOG8-13.xlsx		Feuilles : Simul et Fig2		
	A	B	C	D
1		Nombre	Nombre total	Durée
2	N°	de lancers	de tirs	(en sec.)
3		35	24	333,1
4	1	34	26	335,3
5	2	32	24	317,7
6	3	36	28	339,5
7	4	31	25	295,0
8	5	36	24	346,2
1002	999	34	25	340,0
1003	1000	32	26	310,5

Cellule	Formule
B3	=50-NB.VIDE(SéanceT!B4:B53)
C3	=100-NB.VIDE(SéanceT!G4:G53)-NB.VIDE(SéanceT!J4:J53)
D3	=MAX(SéanceT!L4:L53)

La figure 3 résume les résultats de notre simulation. Le tireur met en moyenne 346,2 secondes pour abattre 20 pigeons, mais une séance peut se prolonger jusqu'à 432,3 secondes, soit un peu plus de 7 minutes et 12 secondes. En moyenne, 35,3 pigeons sont lancés et il tire 26,3 fois. Dans un cas, 43 pigeons ont été nécessaires (voir C7). Une autre fois, il lui a fallu 35 tirs. Noter que ces deux situations extrêmes ne sont pas produites en même temps : lors de l'unique séance où le maximum de 35 tirs a été atteint (n° 138, ligne 141 de la feuille Fig2), seulement 40 pigeons ont été lancés.

Fichier : MOG8-13.xlsx		Feuilles : Résultats et Fig3		
	A	B	C	D
1		Nombre	Nombre total	Durée
2		de lancers	de tirs	(en sec.)
3	Minimum	27	21	279,2
4	5 <sup>e</sup> centile	31	23	306,6
5	Médiane	35	26	345,4
6	95 <sup>e</sup> centile	40	31	392,0
7	Maximum	43	35	432,3
8	Moyenne $\mu$	35,3	26,3	346,2
9	Écart type	2,7	2,5	25,8
10	Intervalle à 90% pour $\mu$ : B.Inf	35,1	26,2	344,8
11	Intervalle à 90% pour $\mu$ : B.Sup	35,4	26,4	347,5

## 14. Les rendez-vous de l'ophtalmologiste

(a) Dans un premier temps, nous analysons une journée typique. La figure 1 indique comment nous avons procédé. Les données des colonnes C, F et I représentent le temps (en minutes) écoulé depuis 8h15, le début de la journée de travail de la spécialiste : par exemple, le ligne 11 (numéro 8) correspond à la période 10h00-10h15 qui commence  $(8 - 1) \times 15 = 105$  minutes après 8h15; en principe, le patient arrive 10 minutes plus tôt, à 9h50, soit  $105 - 10 = 95$  minutes après le démarrage de notre «horloge» et c'est pourquoi nous avons reporté la valeur 95 en C11. Le dernier rendez-vous est fixé à 12h45, soit 4h30 après le tout premier qui est à 8h15; par conséquent, une journée de travail du médecin comprend un maximum de  $1 + (4 \times 4) + 2 = 19$  consultations.

Fichier : MOG8-14a.xlsx					Feuilles : JourT et Fig1				
	A	B	C	D	E	F	G	H	I
1		Le patient		Patient	Attente	Consultation			
2	N°	se présente?	Arrivée	renvoyé?	(en min.)	Début	NPA	Durée	Fin
3	0								0,0
4	1	1	-10	0	10,0	0,0	0,509	17,9	17,9
5	2	0							17,9
6	3	1	20	0	0,0	20,0	0,316	16,1	36,1
7	4	1	35	0	1,1	36,1	0,054	13,6	49,7
8	5	1	50	0	0,0	50,0	0,443	17,2	67,2
9	6	1	65	0	2,2	67,2	0,913	20,0	87,2
10	7	1	80	0	7,2	87,2	0,890	20,0	107,2
11	8	1	95	0	12,2	107,2	0,937	20,0	127,2
12	9	1	110	0	17,2	127,2	0,736	20,0	147,2
13	10	1	125	0	22,2	147,2	0,845	20,0	167,2
14	11	1	140	0	27,2	167,2	0,217	15,2	182,5
15	12	1	155	0	27,5	182,5	0,690	20,0	202,4
16	13	1	170	0	32,4	202,4	0,927	20,0	222,4
17	14	1	185	0	37,4	222,4	0,856	20,0	242,4
18	15	1	200	0	42,4	242,4	0,644	19,4	261,8
19	16	0							261,8
20	17	1	230	0	31,8	261,8	0,503	17,8	279,7
21	18	1	245	0	34,7	279,7	0,588	18,7	298,4
22	19	1	260	1					298,4

Cellule	Formule	Copiée dans :
B4	=SI(ALEA())<=p;1;0)	B5:B22
C4	=SI(B4=0;"";15*A4-25)	C5:C22
D4	=SI(B4=0;"";(I3>290)+0)	D5:D22
E4	=SI(OU(B4=0;D4=1)"";F4-C4)	E5:E22
F4	=SI(OU(B4=0;D4=1)"";MAX(I3;C4))	F5:F22
G4	=SI(OU(B4=0;D4=1)"";ALEA())	G5:G22
I4	=SI(OU(B4=0;D4=1);I3;F4+H4)	I5:I22

Considérons, à titre d'exemple, la ligne 4, qui correspond au 1<sup>er</sup> patient. Nous traiterons les colonnes dans l'ordre B, C, D, G, H, I, E et F.

- En B4, nous testons si ce dernier se présente. Il s'agit de générer un nombre pseudo-aléatoire entre 0 et 1, et de déclarer que le patient vient au rendez-vous quand ce nombre est inférieur

ou égal à  $p=0,9$ . Dans le cas négatif, la consultation ne pourra évidemment avoir lieu et la plage C4:H4 restera vide; les formules des différentes cellules de cette plage incorporent donc un test qui renvoie une cellule vide lorsque B4 contient 0.

- En C4, le test mentionné ci-dessus prend la forme «=SI(B4=0;"";...)». Si  $B4 \neq 0$ , le temps d'arrivée du 1<sup>er</sup> patient se calcule ainsi :  $(15 \times 1) - 25 = -10$ . Ce temps d'arrivée est augmenté de 15 minutes lorsqu'on passe d'une ligne à la suivante, ce qui traduit bien la politique de la spécialiste d'espacer ses rendez-vous d'un quart-d'heure.
- Toujours dans le contexte où  $B4 \neq 0$ , il faut ensuite tester si le patient est renvoyé ou non. On pourrait prétendre que les premiers patients seront sûrement acceptés et qu'effectuer le test seulement dans les dernières lignes suffirait amplement. Mais, où placer la frontière, à quelle période démarrer la procédure de test? Pour simplifier, et aussi pour que les formules dans une colonne soient toutes semblables et s'obtiennent en copiant celles de la ligne 4, nous convenons de faire le test systématiquement pour tous les patients. Il s'agit en fait de vérifier si le temps à l'horloge I sur la ligne précédente, qui représente le moment où l'ophtalmologiste sera disponible pour rencontrer ce patient, est supérieur à 290 (à 13h05, le moment à partir duquel les gens sont renvoyés, il s'est écoulé 290 minutes depuis le démarrage de notre horloge à 8h15). Dans la formule en D4, nous avons ajouté le terme « + 0 » afin que la conclusion du test s'affiche sous la forme 1 ou 0, et non vrai ou faux.
- Si le patient est renvoyé (1 en colonne D), le reste de la ligne est vide, à l'exception de la colonne I. Sinon, il nous faudra simuler la consultation. C'est pourquoi les formules des colonnes E à G commencent par la fonction SI d'Excel :

$$=SI(OU(B4=0;D4=1);"";...)$$

Si le client ne se présente pas ( $B4 = 0$ ) ou est refusé ( $D4 = 1$ ), la formule renvoie une cellule vide. Sinon, un certain calcul sera fait...

- Nous nous intéressons maintenant à la consultation proprement dite. Pour obtenir sa durée, nous générons en colonne G un nombre pseudo-aléatoire entre 0 et 1, que nous transformons en colonne H en une valeur d'une loi triangulaire tronquée à l'aide d'une version adaptée au présent contexte de celle utilisée en section 8.2.4 (voir figure 8.19, colonne G). Voici la formule qui apparaît en H4 et qui sera copiée dans la plage G5:G22 :

$$\text{MIN}(20;\text{SI}(G4 <= X_k; X_{\text{Min}} + (X_{\text{Max}} - X_{\text{Min}}) * \text{RACINE}(X_k * G4); X_{\text{Max}} - (X_{\text{Max}} - X_{\text{Min}}) * \text{RACINE}((1 - X_k) * (1 - G4))))$$

où  $X_k = (X_{\text{Mod}} - X_{\text{Min}}) / (X_{\text{Max}} - X_{\text{Min}}) = (15 - 12) / (28 - 12) = 0,1875$ .

- La colonne I contient une «horloge» indiquant quand la spécialiste sera disponible pour le prochain patient : le temps reporté sur une ligne correspond au moment où se termine la consultation quand il y en a une, et au temps affiché en I sur la ligne précédente dans le cas contraire; évidemment, le temps de fin d'une consultation est égal à la somme du temps de début (colonne F) et de la durée (colonne H). Illustrons par deux exemples tirés de la figure 1 comment se calcule le moment où la spécialiste sera disponible pour le prochain patient :



- ligne 4 : la formule en I4 retourne la valeur 17,9 qui est la somme du temps ( $F4 = 0$ ) où débute la consultation du 1<sup>er</sup> patient et de la durée ( $H4 = 17,9$ ) de cette consultation;
- ligne 5 : comme le 2<sup>e</sup> patient ne s'est pas présenté, le moment où la spécialiste devient disponible reste le même que sur la ligne précédente, soit 17,9.
- L'attente d'un patient (colonne E) est l'écart entre son arrivée (colonne C) et le début de sa consultation (colonne F).
- Il nous reste à indiquer comment obtenir les valeurs en colonne F. Deux cas peuvent survenir.
  - Si le médecin est disponible lorsque le patient arrive, le début de la consultation coïncide avec l'heure d'arrivée du patient. Par exemple, le 3<sup>e</sup> patient arrive à l'instant  $t = 20$  (voir C6); comme la spécialiste est libre depuis l'instant  $t = 17,9$  (voir I5), elle rencontrera le patient sans délai et c'est la valeur  $t = 20$  affichée en C6 qui est reportée en F6.
  - Par contre, le 4<sup>e</sup> patient arrive à l'instant  $t = 35$  (voir C7) et devra attendre que le médecin se libère; cette fois c'est la valeur  $t = 36,1$  affichée en I6 qui est reportée en F7.

Nous avons résumé les résultats de la figure 1 dans la ligne 3 de la feuille Fig2, puis nous avons généré 1000 journées à l'aide de la commande Table de données... d'Excel. Les résultats obtenus sont reproduits dans le figure 2 ci-dessous.

Fichier : MOG8-14a.xlsx			Feuilles : Simul et Fig2			
	A	B	C	D	E	F
1		Nombre de patients		Temps d'attente (en min.)		
2	N°	traités	refusés	moyenne	patient8	patient16
3		16	1	19,1	12,2	
4	1	17	1	16,9	12,5	28,7
5	2	16	0	13,3	7,3	
6	3	18	0	24,1	28,3	22,3
7	4	17	2	28,1	24,5	48,4
1002	999	16	0	20,8	32,0	21,0
1003	1000	16	0	9,9	15,2	1,9

Cellule	Formule
B3	=19-NB.VIDE(JourT!H4:H22)
C3	=SOMME(JourT!D4:D22)
D3	=MOYENNE(JourT!E4:E22)
E3	=JourT!E11
F3	=JourT!E19

Enfin, la figure 3 (voir page suivante) donne les principales statistiques de notre échantillon de 1000 journées. La spécialiste traite en moyenne 16,5 personnes par jour et celles-ci attendent 18,8 minutes en moyenne avant le début de leur consultation. Cependant, pour plus de 5% des journées, l'attente moyenne excède une demi-heure (voir D6). On peut comprendre les récriminations des plus impatientes des patients... et l'intérêt pour la clinique de réviser cette politique de rendez-vous.

Fichier : MOG8-14a.xlsx			Feuilles : Résultats et Fig3			
	A	B	C	D	E	F
1		Nombre de patients		Temps d'attente (en min.)		
2		traités	refusés	moyenne	patient8	patient16
3	Minimum	12	0	2,5	0,0	0,0
4	5 <sup>e</sup> centile	15	0	5,9	0,0	1,5
5	Médiane	17	0	18,6	17,1	26,5
6	95 <sup>e</sup> centile	18	2	33,5	34,6	54,4
7	Maximum	19	3	40,9	43,1	69,4
8	Moyenne $\mu$	16,5	0,63	18,8	17,8	26,7
9	Écart type	1,0	0,80	8,8	11,0	16,3
10	Intervalle à 90% pour $\mu$ : B.Inf	16,4	0,59	18,4	17,2	25,9
11	Intervalle à 90% pour $\mu$ : B.Sup	16,5	0,67	19,3	18,4	27,6

(b) Nous reprenons l'approche appliquée dans la question précédente, en apportant les légers ajustement requis. Ainsi, la feuille Fig4, où est analysée une journée typique, comprend maintenant 15 rendez-vous. De plus, les formules des colonnes C, D et H sont modifiées pour tenir compte du changement de politique : en C4, le coefficient de A4 est 20, et non pas 15, puisque l'intervalle entre deux rendez-vous serait de 20 minutes; le point de comparaison en D4 pour renvoyer un patient est 1h00, soit 300 minutes après le démarrage de notre horloge à 8h00; enfin, la durée de la consultation n'est plus limitée à 20 minutes et la formule de la colonne H ne fait plus intervenir la fonction Min d'Excel.

Fichier : MOG8-14b.xlsx					Feuilles : JourT et Fig4				
	A	B	C	D	E	F	G	H	I
1		Le patient		Patient	Attente	Consultation			
2	N <sup>o</sup>	se présente?	Arrivée	renvoyé?	(en min.)	Début	NPA	Durée	Fin
3									0,0
4	1	1	-10	0	10,0	0,0	0,689	20,0	20,0
5	2	0							20,0
6	3	1	30	0	0,0	30,0	0,439	17,2	47,2
7	4	1	50	0	0,0	50,0	0,608	19,0	69,0
8	5	1	70	0	0,0	70,0	0,061	13,7	83,7
9	6	1	90	0	0,0	90,0	0,204	15,1	105,1
10	7	1	110	0	0,0	110,0	0,828	22,0	132,0
11	8	1	130	0	2,0	132,0	0,258	15,6	147,6
12	9	0							147,6
13	10	1	170	0	0,0	170,0	0,172	14,9	184,9
14	11	1	190	0	0,0	190,0	0,897	23,4	213,4
15	12	1	210	0	3,4	213,4	0,111	14,3	227,7
16	13	1	230	0	0,0	230,0	0,547	18,3	248,3
17	14	1	250	0	0,0	250,0	0,054	13,6	263,6
18	15	1	270	0	0,0	270,0	0,630	19,2	289,2

Cellule	Formule	Copiée dans :
C4	=SI(B4=0;"";20*A4-30)	C5:C22
D4	=SI(B4=0;"";(I3>300)+0)	D5:D22

La figure 5 ci-dessous résume les résultats obtenus. On constate que, dans notre échantillon, personne ne serait renvoyé; que la spécialiste traiterait en moyenne 13,3 patients par jour; que ceux-ci attendraient 3,5 minutes en moyenne.

Fichier : MOG8-14b.xlsx			Feuilles : Résultats et Fig5			
	A	B	C	D	E	F
1		Nombre de patients		Temps d'attente		
2		traités	refusés	moyenne	patient7	patient13
3	Minimum	10	0	0,0	0,0	0,0
4	5 <sup>e</sup> centile	12	0	1,1	0,0	0,0
5	Médiane	13	0	2,9	0,0	0,0
6	95 <sup>e</sup> centile	14	0	8,4	12,3	9,0
7	Maximum	14	0	16,9	25,9	23,9
8	Moyenne $\mu$	13,3	0	3,5	2,8	2,0
9	Écart type	0,8	0	2,4	4,4	3,4
10	Intervalle à 90% pour $\mu$ : B.Inf	13,2	0	3,4	2,6	1,8
11	Intervalle à 90% pour $\mu$ : B.Sup	13,3	0	3,6	3,0	2,2

## 15. Coïncidence d'anniversaires dans un groupe

(a) La figure 1 (voir page suivante) indique comment nous avons procédé pour tester si, parmi  $n$  personnes choisies au hasard, au moins deux ont la même date d'anniversaire. La taille  $n$  du groupe considéré est représentée dans notre fichier MOG8-15a.xlsx par le paramètre TailleG.

Les trois formules commencent par un test, car nous voulions utiliser la même feuille pour traiter des groupes de 20 à 25 personnes. Le test renvoie une cellule vide quand le numéro en colonne A dépasse la taille du groupe. Ainsi, dans la figure 1 où le paramètre TailleG a été fixé à la valeur 20, les 5 dernières lignes n'affichent que le numéro, tandis que les 20 premières simulent chacune un individu différent.

La formule de la colonne B renvoie un entier entre 1 et 365 quand la condition du test est satisfaite. En effet, « 365\*ALEA() » génère un nombre réel pseudo-aléatoire entre 0 et 365, cette dernière valeur étant exclue; la fonction ENT arrondissant à l'entier immédiatement inférieur, « ENT(365\*ALEA()) » retourne un nombre entier entre 0 et 364; le résultat final, après le décalage dû au terme « +1 », se situe entre 1 et 365 tel que désiré. De plus, les 365 valeurs possibles sont équiprobables.

La fonction RANG(B3; B\$3:B\$27;1) commande à Excel de trier en ordre croissant les nombres apparaissant dans la plage B3:B27, puis de renvoyer le rang de celui en B3. Par exemple, la plus petite valeur dans la colonne B de la figure 1 est 39 (voir cellule B20) et on retrouve le rang 1 dans la cellule C20; de même, 44 vient au 2<sup>e</sup> rang (voir ligne 7). Un nombre qui revient plusieurs fois dans la plage se voit attribuer le même rang lors de toutes ses occurrences : ainsi, 94 apparaît deux fois (voir B5 et B10) et le rang 4 est reporté dans les cellules C5 et C10; noter que 107, la valeur suivante de la liste, vient au 6<sup>e</sup> rang, et non au 5<sup>e</sup> (voir C19).

Enfin, la formule `NB.SI(C$3:C$27;A3)` compte le nombre d'occurrences dans la plage C3:C27 de la valeur en A3. Ainsi, le rang 1 apparaît une seule fois en colonne C et Excel renvoie 1 en D3. Par contre, le rang 4 est partagé par deux journées de notre échantillon (voir lignes 5 et 10) et la valeur 2 est affichée en D6.

Fichier : MOG8-15a.xlsx		Feuilles : GroupeT et Fig1		
	A	B	C	D
1		Anniversaire		
2	N°	Jour	Rang	Coïncidence?
3	1	363	20	1
4	2	277	16	1
5	3	94	4	1
6	4	57	3	2
7	5	44	2	0
8	6	296	17	1
9	7	164	11	1
10	8	94	4	1
11	9	202	13	1
12	10	332	18	1
13	11	154	9	1
14	12	138	7	1
15	13	146	8	1
16	14	247	14	1
17	15	358	19	1
18	16	159	10	1
19	17	107	6	1
20	18	39	1	1
21	19	254	15	1
22	20	165	12	1
23	21			
24	22			
25	23			
26	24			
27	25			

Cellule	Formule	Copiée dans :
B3	<code>=SI(A3&lt;=TailleG;1+ENT(365*ALEA());"")</code>	B4:B27
C3	<code>=SI(A3&lt;=TailleG;RANG(B3;B\$3:B\$27;1);"")</code>	C4:C27
D3	<code>=SI(A3&lt;=TailleG;NB.SI(C\$3:C\$27;A3);"")</code>	D4:D27

La cellule A5 de la figure 2 (voir page suivante) indique une façon de tester si au moins deux dates d'anniversaire coïncident dans un groupe donné : il suffit en effet de vérifier si on retrouve un nombre supérieur à 1 dans la colonne D. Pour le groupe représenté à la figure 1, la présence de la valeur 2 dans la cellule D6 permet de conclure que les 2 personnes qui occupent le rang 4

dans ce groupe partagent la même date d'anniversaire. Par contre, si les  $n$  individus du groupe ont tous des dates d'anniversaire différentes, la colonne D de la feuille GroupeT contiendra seulement des « 1 ».

Fichier : MOG8-15a.xlsx				Feuilles : Simul et Fig2			
	A	B	C	D	E	F	G
4		Au moins 1 coïncidence dans le groupe					
5		20	21	22	23	24	25
6	1	1	1	0	1	0	0
7	2	1	1	1	0	0	1
8	3	1	1	0	1	0	1
9	4	0	0	1	1	0	0
5004	4999	0	1	1	0	1	1
5005	5000	0	0	0	0	0	1
Cellule	Formule						
A5	=SI(MAX(GroupeT!D\$3:D\$27)>1;1;0)						

La simulation de la figure 2 a été produite à l'aide de la commande Table de données... d'Excel : nous avons d'abord sélectionné la plage A5:G5005; puis, nous avons défini de la façon suivante les paramètres de cette commande :

- Cellule d'entrée en ligne: \$E\$1 (cette cellule contient la valeur du paramètre TailleG qui représente la taille  $n$  du groupe).
- Cellule d'entrée en colonne: \$A\$5.

La tableau qui suit résume les résultats obtenus. On constate que, dans nos échantillons, la proportion des groupes où se produit au moins une coïncidence dépasse 50% quand celui-ci comprend 23 personnes ou plus. Ce qui confirme l'affirmation de l'énoncé.

Taille	20	21	22	23	24	25
Proportion	0,4070	0,4420	0,4778	0,5144	0,5360	0,5662

**Note.** Voici comment calculer de façon exacte la probabilité  $p_n$  qu'il y ait au moins une coïncidence dans un groupe de taille  $n$ , sous l'hypothèse que les gens sont choisis de façon aléatoire et que les 365 jours de l'année sont équiprobables :

$$1 - p_n = P(n \text{ dates différentes}) = \frac{\binom{365}{n} \times n!}{365^n} = \frac{365! \times n!}{365^n \times n! \times (365-n)!} = \frac{365!}{365^n \times (365-n)!}.$$

Le tableau ci-dessous donne les valeurs de  $p_n$ . L'écart entre la probabilité  $p_n$  et l'estimé fourni par l'échantillon est inférieur à 0,5% dans tous les cas, sauf lorsque  $n = 23$  et alors l'écart est d'environ 0,7%.

Taille	20	21	22	23	24	25
$p_n$	0,4114	0,4437	0,4757	0,5073	0,5383	0,5687

(b) Il suffit de répéter les calculs effectués dans la question précédente, mais en fixant la taille du groupe aux valeurs 35, 36, 40 et 41. Voici les résultats de la simulation.

Taille	35	36	40	41
Proportion	0,819	0,830	0,882	0,898

Le professeur propose aux étudiants un pari qui est fortement biaisé en sa faveur. On pourrait s'interroger sur son sens de l'éthique... à moins que les sommes en jeu soient si faibles que le côté pédagogique prime sur l'aspect financier.

(c) Il suffit de répéter les calculs effectués dans la question (a), mais en faisant varier la taille du groupe de 85 à 90 et en modifiant ainsi la formule de la cellule A5 de la feuille Simul :

$$=SI(MAX(GroupeT!D\$3:D\$92)>2;1;0)$$

Voici les résultats de la simulation.

Taille	85	86	87	88	89	90
Proportion	0,4515	0,4790	0,4900	0,5075	0,5145	0,5310

## 16. Les machines à sous.

(a) Certaines plages de la feuille Données, auxquelles nous référons dans les formules apparaissant dans les autres feuilles, se sont vus attribuer des noms :

- dans les simulations, le numéro de la facette d'un disque sera noté  $X$ ; la liste des valeurs de cette variable aléatoire, qui occupent les cellules B11 à B17, sera nommée ValeursX;
- la plage B11:C17 donne les prix (en \$) lorsque les trois disques affichent le même symbole et porte le nom Prix3;
- Cumul1, Cumul2 et Cumul3 réfèrent aux plages E11:E17, G11:G17 et I11:I17 respectivement.

Fichier : MOG8-16.xlsx				Feuilles : Données et Fig1					
	A	B	C	D	E	F	G	H	I
8	<b>Répartition des symboles sur les facettes des trois disques et prix pour 3 symboles identiques</b>								
9			Prix	Disque de gauche		Disque central		Disque de droite	
10	Code	N°	(en \$)	Nombre	$P(X1 < x)$	Nombre	$P(X2 < x)$	Nombre	$P(X3 < x)$
11	Globe	1	500	1	0,000	1	0,000	1	0,000
12	Barre	2	100	1	0,045	2	0,045	1	0,045
13	Prune	3	50	3	0,091	3	0,136	1	0,091
14	Cloche	4	20	2	0,227	4	0,273	4	0,136
15	Orange	5	15	5	0,318	2	0,455	5	0,318
16	Cerise	6	10	3	0,545	2	0,545	2	0,545
17	Citron	7	0	7	0,682	8	0,636	8	0,636
18	Nombre de facettes			22		22		22	
Cellule	Formule						Copiée dans :		
D18	=SOMME(D11:D17)						F18, H18		
E12	=E11+D11/D\$18						E13:E17, G13:G17, I13:I17		

Dans un premier temps, nous simulons la séance d'un joueur typique. La figure 2 qui suit indique comment nous avons procédé. Chacune des lignes 3 à 202 représente l'un des 200 coups. Les colonnes B, C et D indiquent quels symboles apparaissaient sur chacun des trois disques : par exemple, la formule en B3 commande de générer un nombre pseudo-aléatoire entre 0 et 1, puis de le transformer en un numéro de 1 à 7 en utilisant les fréquences cumulées de la plage E11:E17 de la feuille Données. Ainsi, dans le cas illustré à la figure 2, la fonction ALEA() a renvoyé un nombre entre 0,545 et 0,636; cette valeur correspondant au numéro 6, le disque 1 affichait une cerise lors de ce 1<sup>er</sup> coup. La formule de la colonne E donne le nombre de cerises pour le coup associé à la ligne; plus techniquement, il s'agit de compter combien de fois apparaît le numéro 6 dans les colonnes B, C et D de cette ligne.

Enfin, la cellule F3 calcule le prix remporté par le joueur lors de ce coup à l'aide de la formule :

=SI(E3<>0;CHOISIR(E3;2;5;20);SI(ET(B3=C3;C3=D3;B3<>7);RECHERCHEV(B3;Prix3;2;0);""))

Deux cas sont considérés.

- Si au moins une cerise a été obtenue, autrement dit si la valeur en colonne E de cette ligne est non nulle, on choisit entre des prix de 2\$, 5\$ ou 20\$ selon que le nombre de cerises est 1, 2 ou 3.
- Sinon, on teste si les trois disques affichent le même symbole. Dans le cas négatif, le joueur ne gagne pas ce coup-là et on ne reporte rien en colonne F. Dans le cas contraire, la formule renvoie le montant du prix, pourvu que le symbole commun ne soit pas un citron. Noter qu'aucun prix n'est accordé quand le joueur obtient trois citrons et que la formule ne renvoie rien dans ce cas.

Fichier : MOG8-16.xlsx				Feuilles : JoueurT et Fig2		
	A	B	C	D	E	F
1	Essai	Résultat			Nombre de	Prix
2	N°	disque 1	disque 2	disque 3	cerises	(en \$)
3	1	6	7	7	1	2
4	2	5	7	2	0	
5	3	4	7	4	0	
6	4	6	5	6	2	5
7	5	5	5	7	0	
8	6	3	5	5	0	
9	7	7	6	7	1	2
10	8	5	4	7	0	
11	9	2	2	5	0	
200	198	5	7	7	0	
201	199	7	6	6	2	5
202	200	6	5	7	1	2

Cellule	Formule	Copiée dans :
B3	=RECHERCHE(ALEA();Cumul1;ValeursX)	B4:B202
E3	=(B3=6)+(C3=6)+(D3=6)	E4:E202
F3	=SI(E3<>0;CHOISIR(E3;2;5;20);...)	F4:F202

La ligne 3 de la figure 3 (voir page suivante) résume les résultats de la séance illustrée à la figure 2. La formule de B3 compte le nombre de prix récoltés par le joueur en 200 coups : en effet, le fait de ne pas gagner lors d'un coup donné se traduit par une cellule vide dans la

colonne F; le nombre de gains pendant une séance de 200 coups s'obtient donc en retranchant de 200 le nombre de cellules vides de la plage F3:F202 de la feuille JoueurT. La cellule C3 s'intéresse, elle, à la valeur des prix : plus précisément, elle calcule la somme des prix reçus par le joueur pendant la séance. Enfin, la cellule D3 donne le revenu net du joueur : celui-ci s'obtient en déduisant de son revenu (C3) la somme investie, qui est égale au produit du nombre de coups (200) par la mise pour un coup (1\$).

Fichier : MOG8-16.xlsx		Feuilles : SimulA et Fig3		
	A	B	C	D
1	Joueur	Nombre	Somme totale	Revenu net
2	N <sup>o</sup>	de prix	reçue en prix	(ou perte nette)
3		69	194	-6 \$
4	1	47	137	-63 \$
5	2	54	117	-83 \$
6	3	58	169	-31 \$
7	4	66	208	8 \$
8	5	56	137	-63 \$
1002	999	59	160	-40 \$
1003	1000	59	175	-25 \$
Cellule Formule				
B3 =200-NB.VIDE(JoueurT!F3:F202)				
C3 =SOMME(JoueurT!F3:F202)				
D3 =C3-200*1				

La séance typique est ensuite répétée 1000 fois à l'aide de la commande Table de données... d'Excel. Les résultats obtenus sont reproduits dans le figure suivante. On note, sans surprise, que la machine mérite son surnom de «bandit manchot». En effet, dans l'échantillon considéré ici, le revenu net médian (D3) s'établit à -31\$; de plus, les joueurs perdent 15,20\$ en moyenne (D8). Cependant, quelques chanceux réussissent à sortir gagnants de la séance: par exemple, l'un des 1000 joueurs a complété la séance avec des gains nets de 539\$.

Fichier : MOG8-16.xlsx		Feuilles : RésultatA et Fig4		
	A	B	C	D
1		Nombre	Somme totale	Revenu net
2		de prix	reçue en prix	(ou perte nette)
3	Minimum	39	93	-107
4	5 <sup>e</sup> centile	48	123	-77
5	Médiane	59	169	-31
6	95 <sup>e</sup> centile	69	261	61
7	Maximum	77	739	539
8	Moyenne $\mu$	58,9	184,8	-15,20
9	Écart type	6,3	80,7	80,67
10	Intervalle à 90% pour $\mu$ : B.Inf	58,6	180,6	-19,40
11	Intervalle à 90% pour $\mu$ : B.Sup	59,2	189,0	-11,00

(b) Les pertes des joueurs se traduisent en gains pour le propriétaire : ainsi, pour l'échantillon illustré considéré à la question (a), le revenu net moyen du propriétaire est de 15,20\$ par joueur.



(c) La figure 5 illustre une soirée où la séance de Charles a duré 640 coups. Pour le 1<sup>er</sup> coup (ligne 3), nous avons copié les formules de la figure 2 dans la plage B3:F3. Mais, à partir du 2<sup>e</sup> coup, nous avons ajouté un test pour vérifier si la séance s'est poursuivie jusque-là. Plus techniquement, il s'agit de tester la valeur du solde sur la ligne précédente : s'il est nul, la séance est terminée et les formules des colonnes B à F renvoient des cellules vides; sinon, Charles continue à jouer et nous reprenons les formules de la ligne 3.

Deux cas sont à considérer pour le calcul du solde de la colonne G.

- Une fois que le solde est tombé à 0\$, la séance est terminée et le solde sur les lignes subséquentes sera nul.
- En cours de séance, le solde s'obtient en retranchant du solde précédent la mise de 1\$, puis en ajoutant le prix éventuel reporté en colonne F. (Comme Excel renvoie un message d'erreur quand on lui commande d'additionner une cellule numérique et une cellule vide, nous avons inclus un test sur le contenu de la cellule en colonne F : si celle-ci est vide, le joueur n'a pas reçu de prix lors de ce coup et le terme représentant le prix sera le nombre 0.)

Fichier : MOG8-16.xlsx				Feuilles : SoirT et Fig5			
	A	B	C	D	E	F	G
1	Essai	Résultat			Nombre de	Prix	Solde
2	N <sup>o</sup>	disque 1	disque 2	disque 3	cerises	(en \$)	(en \$)
3	1	2	6	5	1	2	21
4	2	3	3	3	0	50	70
5	3	2	1	6	1	2	71
6	4	7	7	4	0		70
7	5	7	7	5	0		69
8	6	5	7	6	1	2	70
9	7	4	4	1	0		69
10	8	1	7	6	1	2	70
640	638	5	3	1	0		2
641	639	5	7	7	0		1
642	640	5	7	5	0		0
Cellule	Formule					Copiée dans :	
B4	=SI(G3=0;"";RECHERCHE(ALEA());Cumul1;ValeursX))					B5:B1002	
E4	=SI(G3=0;"";(B4=6)+(C4=6)+(D4=6))					E5:E1002	
F4	=SI(G3=0;"";SI(E4<>0;CHOISIR(E4;2;5;20);...))					F5:F1002	
G4	=SI(G3=0;0;G3-1+SI(F4=""0;+F4))					G5:G1002	

Nous avons réservé 2000 lignes pour les séances. Cette valeur est parfois insuffisante : en effet, comme le montre la figure 6 (voir page suivante), lors de la soirée numéro 997, Charles possédait 306\$ après 2000 coups. L'allongement considérable de certaines séances provient des prix de 500\$ qui, chacun, assurent au joueur 500 coups additionnels.

Noter que la formule de B3 indique combien de coups ont été nécessaires ce soir-là pour épuiser les 20\$ investis; que celle de C3 donne le solde final entre les mains de Charles après 2000 coups lorsque la séance dépasse cette limite.

Fichier : MOG8-16.xlsx Feuilles : SimulC et Fig6			
	A	B	C
1	Soir	Nombre	Solde final
2	N°	de coups	(en \$)
3		640	0
4	1	117	0
5	2	629	0
6	3	109	0
996	993	284	0
997	994	2001	306
998	995	73	0
999	996	52	0
1000	997	34	0
1001	998	62	0
1002	999	63	0
1003	1000	969	0
Cellule	Formule		
B3	=2000-NB.SI(Soirt!G3:G2002;0)+1		
C3	=Soirt!G2002		

La figure 7 résume les résultats de la simulation de la figure précédente. En moyenne, Charles aurait joué 173 coups par soir (B13). Mais, à 2 reprises, il n'aurait gagné aucun prix (B2) et sa séance se serait terminée dès le 20<sup>e</sup> coup. Par contre, 22 séances se seraient prolongées au-delà du 2000<sup>e</sup> coup (B4). Si Charles avait dû cesser après ce 2000<sup>e</sup> coup – par exemple, parce que le bar fermait à ce moment-là –, il aurait dépensé en moyenne  $20 - 9,41 = 10,59$  dollars (B5); un soir, il aurait même quitté avec 903\$ en poche (B6).

Fichier : MOG8-16.xlsx Feuilles : RésultC et Fig7		
	A	B
1	Cas extrêmes	
2	Nombre de soirées à 20 coups	2
3	Nombre de soirées à plus de 1000 coups	33
4	Nombre de soirées à plus de 2000 coups	22
5	Solde final moyen	9,41 \$
6	Solde final maximal	903 \$
7	Caractéristiques de la variable « Nombre de coups »	
8	Minimum	20
9	5 <sup>e</sup> centile	29
10	Médiane	70
11	95 <sup>e</sup> centile	665
12	Maximum	2001
13	Moyenne $\mu$	173,1
14	Écart type	335,5
15	Intervalle à 90% pour $\mu$ : B.Inf	155,7
16	Intervalle à 90% pour $\mu$ : B.Sup	190,6

**Note.** On peut montrer théoriquement (voir fichier MOG8-16.xlsx, feuille Prob) que la perte attendue après 200 coups est de 18,18\$. De plus, la probabilité que Charles ne gagne aucun prix lors des 20 premiers coups est de 0,09%; on en déduit que la variable  $S$ , «nombre de soirs dans un échantillon de 1000 qui se terminent dès le 20<sup>e</sup> coup», obéit à une loi binomiale  $B(1000; 0,09\%)$  et que  $E(S) = 0,9$ . Enfin, la distribution de  $S$  est résumée dans le tableau suivant.

s	0	1	2	3	4	5	6+
P(S = s)	39,92%	36,68%	16,83%	5,14%	1,18%	0,22%	0,04%

## 17. Rupture de stock et rabais

Nous décrivons ici seulement les modifications à apporter au modèle de gestion des stocks décrit dans la section 8.2.3. La feuille Données contient deux paramètres, en plus de ceux énumérés au tableau 8.6 :

- CRabais = 50 représente le montant (en \$) offert par le magasin en cas de rupture de stock.
- PropRev = 70% est la proportion des clients qui acceptent de revenir lorsqu'on leur propose le rabais.

La figure 1 qui suit reproduit quelques extraits de la feuille Simul où sont calculés les coûts pour une période de 5000 jours consécutifs.

Fichier : MOG8-17.xlsx										Feuilles : Simul et Fig1					
	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1	Jour	Stock	Quantité	Stock	Demande		Stock	Pénurie	Rabais acceptés		Comm.	Jour de	Stock	Coûts	Revenus
2	n°	début	reçue	ouvert.	brute	satisf.	ferm.		Le jour $i$	Cumul	?	réception	au livre	de gestion	nets
3	1	60	0	60	12	12	48	0	0	0			48	13,50	3 286,50
4	2	48	0	48	10	10	38	0	0	0			38	10,75	2 739,25
5	3	38	0	38	10	10	28	0	0	0			28	8,25	2 741,75
6	4	28	0	28	11	11	17	0	0	0			17	5,63	3 019,38
7	5	17	0	17	10	10	7	0	0	0	Oui	8	107	203,00	2 547,00
8	6	7	0	7	11	7	0	4	3	3			97	25,88	2 574,13
9	7	0	0	0	9	0	0	9	5	8			92	100,00	1 025,00
10	8	0	100	92	13	13	79	0	0	0			79	21,38	3 553,63
5001	4999	67	0	67	3	3	64	0	0	0			64	16,38	808,63
5002	5000	64	0	64	4	4	60	0	0	0			60	15,50	1 084,50

Cellule	Formule	Copiée dans :
C7	=TComm*NB.SI(L3:L6;A7)	C8:C5002
D7	=B7+C7-SI(C7=0;0;J6)	D4:D5002
I7	=CRITERE.LOI.BINOMIALE(H7;PropRev;ALEA())	I4:I5002
J7	=SI(C7=0;J6+I7;0)	J4:J5002
M7	=M6-F7+SI(K7="Oui";TComm;0)-I7	M4:M5002
N7	=CStock*(D7+G7)/2+CPénurie*(H7-I7)+CComm*(K7="Oui")	N4:N5002
O7	=MargeU*F7-N7+(MargeU-CRabais)*I7	O4:O5002

Les formules des colonnes B, E, F et G sont identiques à celles du fichier Gestion-Stocks.xlsx (voir figure 8.16) et ne sont pas reproduites ici. Celles de I et J calculent le nombre de rabais acceptés

par les clients et sont commentées ci-dessous. L'inclusion de ces deux colonnes force le décalage des colonnes subséquentes, ce qui entraîne divers ajustements à plusieurs formules de la figure 8.16; dans certains cas – C7, K7 et L7–, la seule différence entre les formules des fichiers Gestion-Stocks.xlsx et MOG8-17.xlsx réside dans les noms des colonnes auxquelles elles réfèrent; à titre d'exemple, nous donnons la version modifiée de celle de C7.

Nous commentons maintenant les changements dans la feuille Simul qui découlent de l'hypothèse qu'en cas de rupture de stock, une proportion des clients acceptent de revenir acheter la batterie en échange d'un rabais.

- Pour simplifier, nous considérons que tous les clients en possession d'un «coupon-rabais» de 50 \$ se présenteront le jour de la réception d'une commande. Nous convenons donc de modifier de la façon suivante la formule du stock à l'ouverture (D7) : dans la figure 8.16, la valeur en D7 se calcule comme la somme des unités restant la veille au soir (G6, repris en B7) et celles reçues le matin même (C7); ici, de ce total, nous retranchons le nombre de coupons-rabais en circulation (J6) quand une commande est reçue ( $C7 \neq 0$ ).
- Le nombre  $C$  de coupons-rabais émis un jour donné (I7) obéit à une loi binomiale dont le paramètre «nombre de tirages» est égal au nombre d'unités en pénurie reporté en H7. La fonction ALEA() de la formule en I7 commande de générer un nombre pseudo-aléatoire entre 0 et 1, que la fonction CRITERE.LOI.BINOMIALE transforme en une valeur de la variable  $C$ .
- Nous avons supposé que tous les clients profitent de leurs coupons-rabais le plus tôt possible. Par conséquent, si une commande est reçue un matin donné, il ne restera aucun coupon en circulation ce soir-là. Plus techniquement, il s'agit d'exiger que J7 soit nul quand  $C7 \neq 0$ . Dans le cas contraire, le nombre cumulé de coupons en circulation un soir donné (J7) est la somme du nombre de coupons en circulation la veille au soir (J6) et du nombre de coupons émis ce jour-là (I7).
- Le stock au livre (M7) inclut les unités réellement en stock, plus les unités commandées et en instance de livraison, moins les unités en pénurie pour lesquelles le magasin a émis un coupon. Nous ajoutons donc le terme «-I7» à la formule utilisée dans le modèle de la section 8.2.3.
- Dans les coûts de gestion, le coût de pénurie s'applique seulement aux clients qui refusent le coupon-rabais de 50\$. C'est pourquoi nous multiplions ici le paramètre CPénurie par (H7-I7).
- Lorsqu'il y a pénurie et que le client accepte de revenir plus tard chercher sa batterie, les revenus de la vente reportée dans le temps sont imputés à la journée où le client est venu pour la 1<sup>re</sup> fois et a pris la décision d'achat. C'est ce qui explique la présence du terme «(MargeU-CRabais)\*I7» dans la formule de la cellule O7.

**Note.** Comme dans le modèle de la section 8.2.3, la cellule B4 et la plage D4:O4 sont copiées dans les lignes 5 à 5002. Mais, c'est seulement à partir de la ligne 7 que la formule de la colonne C est «stable» et peut être copiée.

Comme dans la section 8.2.3, nous poursuivons notre étude en analysant l'impact des paramètres TComm et PComm à l'aide de la commande Table de données... d'Excel. Les résultats obtenus

sont reproduits dans le tableau ci-dessous. On observe, comme dans la section 8.2.3, que les variations de revenu sont, à quelques exceptions près, peu ou pas significatives, pourvu que PComm soit supérieur à 20.

TComm	PComm	Nombre de commandes	Nombre total d'unités		Demande totale	Revenu net moyen
			Pénurie	Stock		
100	10	381	3 037	223 312	38 979	2 043
	15	381	1 699	243 313	38 677	2 056
	20	386	1 022	264 903	38 938	2 087
	25	386	572	288 167	38 725	2 085
	30	385	173	316 659	38 590	2 087
	35	390	48	342 126	38 998	2 111
	40	388	43	357 305	38 756	2 097
	45	388	1	383 445	38 747	2 096
	50	387	2	412 636	38 680	2 091
100	20	388	976	264 113	39 047	2 095
150		257	672	390 270	38 762	2 085
200		192	482	513 196	38 532	2 073
250		154	324	643 149	38 553	2 075
300		130	377	765 471	38 929	2 088
350		111	230	886 460	38 933	2 087
400		97	216	1 014 141	38 906	2 080
450		86	172	1 140 112	38 805	2 069
500		79	330	1 251 045	39 172	2 080

### 18. Les quatre relais d'un panneau de contrôle.

La figure 1 (voir page suivante) analyse 1051 pannes consécutives dans le contexte où la politique de l'entreprise consiste à remplacer seulement le relai défaillant. On notera que les formules en B3 et en B4 diffèrent. En effet, au départ (B3:E3), les quatre relais sont neufs et leur durée de vie s'obtient en générant une valeur pseudo-aléatoire d'une loi uniforme. Mais, par la suite (B4:E1053), seul le relai qui vient d'être remplacé est neuf et il faut modifier la formule pour tenir compte de l'âge des trois autres : par exemple, au moment de la 1<sup>re</sup> panne (ligne 4), le premier relai fonctionne depuis 157,2 h (F3); comme sa durée de vie est de 166,0 h (B3), il lui reste  $166,0 - 157,2 = 8,8$  heures (B4) de survie. De façon générale, pour calculer combien d'heures il reste à un relai donné, deux cas sont à considérer, selon que le relai vient d'être remplacé ou non, que nous illustrerons par deux exemples tirés de la ligne 4.

- En E4, la formule commande d'abord de tester si  $E3 = \$F3$  : comme la réponse est positive, la fonction SI renvoie la 1<sup>re</sup> valeur, qui représente la durée de vie du relai qui vient d'être remplacé.

- En B4, la condition « B3=\$F3 » n'est pas satisfaite et Excel reportera dans la cellule la valeur du 2<sup>e</sup> argument : par conséquent,  $B4 = B3 - \$F3 = 182,3 - 162,1 = 20,2$ . Autrement dit, le relai 1 fonctionnera encore pendant 20,2 heures.

Les données de la colonne F représentent les délais (en heures) entre deux pannes consécutives. Comme la production de l'aciérie s'arrête dès que l'un des quatre relais tombe en panne, ces délais se calculent comme le minimum des durées de survie reportées dans les colonnes B à E.

L'horloge de la colonne G donne l'écart (en heures) entre le début du processus et le moment où se produit une panne. Par exemple, la 2<sup>e</sup> (voir cellule G4) survient à l'instant  $t = 162,1 + 4,8 = 166,9$ . En termes algébriques, le temps reporté dans la cellule G4 s'obtient comme la somme du temps à l'horloge sur la ligne précédente (G3) et du délai (F4) entre les pannes numéros 1 et 2.

Fichier : MOG8-18.xlsx		Feuilles : Pol1 et Fig1					
	A	B	C	D	E	F	G
1		Durée de survie (en h) du relai					
2	N°	Relai 1	Relai 2	Relai 3	Relai 4	Min	Horloge
3	-50	182,3	189,6	166,9	162,1	162,1	162,1
4	-49	20,2	27,5	4,8	177,4	4,8	166,9
5	-48	15,4	22,7	170,2	172,5	15,4	182,3
52	-1	52,1	186,7	57,1	10,6	10,6	2 228,4
53	0	41,6	176,2	46,5	189,5	41,6	2 270,0
54	1	191,4	134,6	5,0	147,9	5,0	2 275,0
55	2	186,4	129,6	150,7	143,0	129,6	2 404,6
1052	999	181,2	89,4	123,6	3,2	3,2	45 979,8
1053	1000	178,0	86,2	120,4	167,8	86,2	46 066,0

Cellule	Formule	Copiée dans :
B3	=XMin+ALEA()*(XMax-XMin)	C3:E3
B4	=SI(B3=\$F3;XMin+ALEA()*(XMax-XMin);B3-\$F3)	B4:E1053
F3	=MIN(B3:E3)	F4:F1053
G3	=F3	-----
G4	=G3+F4	G5:G1053

Si l'entreprise décidait de remplacer les quatre relais après chaque panne, l'analyse précédente prendrait la forme illustrée à la figure 2 (voir page suivante). L'approche est identique à celle de la figure 1, sauf que la formule de B4 n'apparaît pas cette fois et que celle de B3 est copiée dans la plage B3:E1002. En effet, en vertu de la nouvelle politique, le laminoir est muni de quatre relais neufs après chaque panne, ce qui signifie que, sur toutes les lignes, le temps restant de fonctionnement de chacun coïncide avec sa durée de vie et obéit à une loi uniforme.

Fichier : MOG8-18.xlsx		Feuilles : Pol2 et Fig2					
	A	B	C	D	E	F	G
1	N°	Durée de vie (en h) du relai				Min	Horloge
2		Relai 1	Relai 2	Relai 3	Relai 4		
3	1	187,0	163,3	168,1	175,8	163,3	163,3
4	2	152,7	164,7	189,8	186,6	152,7	316,0
5	3	180,2	169,3	184,6	195,7	169,3	485,3
6	4	190,9	186,8	190,7	153,5	153,5	638,8
1001	999	185,4	163,9	168,7	180,9	163,9	160 160,1
1002	1000	157,2	171,4	170,6	164,9	157,2	160 317,3
Cellule	Formule						Copiée dans :
B3	=XMin+ALEA()*(XMax-XMin)						B3:E1002
F3	=MIN(B3:E3)						F4:F1002
G3	=F3						-----
G4	=G3+F4						G5:G1002

Les résultats des deux simulations précédentes sont résumés dans la figure 3 (voir ci-dessous). Les paramètres utilisés CoûtR = 850 et CoûtH = 1000 représentent, le premier, le coût d'un relai et, le second, le manque à gagner encouru par l'entreprise lorsque la production est arrêtée pendant une heure à cause d'une panne. Dans les calculs associés à la politique 1 (colonne C), on ne tient pas compte des lignes 3 à 53 : en effet, au départ, les quatre relais sont neufs simultanément, une situation exceptionnelle; on considère donc que la période couverte par les pannes numérotées de -50 à 0 constitue un temps de démarrage, à ignorer, et que, par la suite, le processus a atteint une allure de croisière qui est représentative.

Fichier : MOG8-18.xlsx		Feuilles : Résultats et Fig3	
	A	B	D
18			Politique 1
19	Horloge (en h)		43 796,00
20	Production (en h/panne)		160,3
21	Coût (en \$/h)		42,24
21			33,68
Cellule	Formule		Copiée dans :
C19	='Pol1'!G\$1053-'Pol1'!G\$53		-----
D19	='Pol2'!G\$1002		-----
C20	=C19/1000		D20
C21	=(CoûtR+CoûtH)*1000/C19		-----
D21	=(4*CoûtR+2*CoûtH)*1000/D19		-----

Les coûts des politiques 1 et 2 sont exprimés en dollars par heure. Dans les deux cas, ils se calculent en multipliant le coût associé à une panne par le nombre de pannes (1000), puis en divisant par la durée de la période (C19 ou D19). Le 1<sup>er</sup> facteur diffère selon la politique retenue.

- Si l'entreprise remplace seulement le relai défectueux, une panne entraîne un déboursé de CoûtR = 850 dollars pour le relai neuf et un manque à gagner de CoûtH = 1000 dollars

découlant de l'arrêt de la production pendant une heure; au total,  $\text{CoûtR} + \text{CoûtH} = 1850$  dollars.

- Si la nouvelle politique est implantée, la production est interrompue pendant 2 heures, ce qui revient à  $(2 \text{ CoûtH})$  dollars, et il faut payer 4 relais neufs; au total,  $(2 \text{ CoûtR}) + (4 \text{ CoûtH}) = 5400$  dollars par panne.
- (a) La politique courante de gestion des pannes revient à 42,24\$/h.
- (b) Si la nouvelle politique était implantée, la gestion des pannes reviendrait à 33,68\$/h.
- (c) La nouvelle politique semble la plus intéressante : d'abord, et surtout, elle coûte bien moins cher; de plus, elle assure une production plus soutenue, plus régulière, ce qui facilite la planification.